
VBA 
(სალექციო კურსი) 



 
   
VBA: 1. მაკროს ჩაწერა    
მაკროს ჩამწერის გამოყენება   
გახსენით Excel–ი და VBE (Visual Basic Editor – VB რედაქტორი). VBE ფანჯარა შეიცავს Project Explorer და Properties 
ფანჯრებს მარცხნივ. Project Explorer: მუშაობს როგორც ფაილ–მენეჯერი, გეხმარებათ თქვენი სამუშაო პროექტის 
კოდის ნავიგაციაში. Properties Window: გიჩვენებთ გააქტიურებული ობიექტის თვისებებს (მაგალითად Sheet1, თუ 
პროექტი Excel–შია ).   
  
სავარჯიშო 1: მაკროს ჩაწერა.  
   
სავარჯიშო გვიჩვენებს თუ რა ხდება მაკროს ჩაწერისას და განსხვავებას აბსოლუტური და ფარდობითი მიმართვების 
ჩაწერებს შორის.   
1. ახალი წიგნის ცარიელ ფურცელზე აირჩიეთ უჯრედი C10   
2. გაუშვით მაკროს ჩამწერი (Macro Recorder) ოფციით Store macro in: This Workbook. VBE შექმნის ახალი მოდულის 
ფოლდერს. შეგვიძლია დავათვალიეროთ იგი – ეს მოქმედებები მაკროში არ ჩაიწერება. დააწკაპუნეთ [+] ნიშანზე 
ფოლდერის გვერდით და დარწმუნდით რომ VBE–მ ნამდვილად ჩაამატა მოდული სახელით Module1. მოდულის 
იარლიყზე ორმაგი დაწკაპუნებით გახსენით მისი კოდის ფანჯარა. გადავიდეთ უკან Excel–ში.   
3. დარწმუნდით რომ ღილაკი Use Relative Reference, რომელიც მდებარეობს View ჩანართის Macros ჯგუფში (ან 
Developer ჩანართის Code ჯგუფში), არ არის ჩართული.   
4. აირჩიეთ უჯრედი B5 და გააჩერეთ მაკროს ჩამწერი.   
5. გადაერთეთ ისევ VBE–ში და შეხედეთ კოდს:   
  
Range("B5").Select   
6. ჩავწეროთ სხვა მაკრო, იგივე გზით ოღონდ ჩართული ღილაკით Use Relative Reference.   
7. გადაერთეთ VBE–ში და შეხედეთ კოდს:   
  
ActiveCell.Offset(-5, -l).Range(“Al”).Select   
8. ახლა კვლავ ჩავწეროთ ახალი მაკრო, მაგრამ B5 უჯრედის არჩევის მაგიერ ავირჩიოთ უჯრედების ბლოკი დაწყებული 
B5–დან  F7–მდე (select cells B5:F7)   
9. გადაერთეთ VBE–ში და შეხედეთ კოდს:   
  
ActiveCell.Offset(-5, -l).Range("Al:E3").Select   
10. გაუშვით მაკროები წინასწარ მონიშნული, C10–საგან განსხვავებული, უჯრედით ( Macro2 და Macro3 
მაკროებისათვის მონიშნული უჯრედი უნდა იყოს მეექვსე სტრიქონის დაბლა - იხ. ნაბიჯი 11)   
Macro1 - ყოველთვის გადაიტანს მონიშვნას B5 უფრედში, Macro2 - გადაიტანს მონიშვნას 5 სტრიქონით ზევით და 1 
სვეტით მარცხნივ წინასწარ მონიშნული უჯრედის მიმართ. Macro3 - ყოველთვის მონიშნავს ექვსი უჯრედისაგან 
შედგენილ ბლოკს დაწყებული 5 სტრიქონით ზევით და 1 სვეტით მარცხნივ მაკროს გაშვებამდე მონიშნული უჯრედის 
მიმართ.   
11. გავუშვათ Macro2 ისე, რომ წინასწარ მოვნიშნოთ უჯრედი პირველ 5 სტრიქონში, ანუ გამოვიწვიოთ შეცდომა. მაკრო 
ცდილობს არ არსებული უჯრედის მონიშვნას და გამოიტანს შეცდომის  შეტყობინებას. დააჭირეთ Debug რათა 
გამოვაჩინოთ მაკროს ის ნაწილი სადაც შეცდომაა.   
  
შენიშვნა: როცა VBE მუშაობს Debug რეჟიმში შეცდომის შემცველი კოდის სტრიქონი გამოყოფილია ყვითელი ფერით. 
ვიდრე რამეს მოიმოქმედებდეთ მაკრო უნდა შეაჩეროთ Reset ღილაკით ან VBE პანელიდან Run –> Reset. ყვითელი 
მონიშვნა გაქრება და VBE გამოვა Debug რეჟიმიდან.   
12.  აქ ყველაზე იოლი გამოსავალია შეცდომის იგნორირება და შემდეგ ინსტრუქციაზე გადასვლა. ჩავწეროთ 
სტრიქონი On Error Resume Next  მაკროს პირველივე სტრიქონის ზევით ( Sub Macro2() სტრიქონის ქვევით).   



13.  გაუშვით Macro2 წინასწარი მონიშვნით პირველ 5 სტრიქონში. ამ შემთხვევაში შეცდომის შეტყობინება არ 
გვექნება. ეს ძალზედ მარტივი მაკროა. მომხმარებელი ვერც გაიგებს რომ აქ რაღაც შეცდომასთან ჰქონდა საქმე.   
14. შევცვალოთ Macro2–ის კოდი ასეთნაირად:  
  
           Sub Macro2()  
             On Error GoTo ErrorHandler   
             ActiveCell.Offset(-5, -l).Range("Al").Select  
             Exit Sub   
  ErrorHandler: MsgBox "You must start below Row 5"  
  End Sub   
15.  ამჯერად მომხმარებელი მიიღებს შეტყობინებას რომ დაშვებულია შეცდომა, კერძოდ, მან უნდა მონიშნოს უჯრედი 
მე–5 სტრიქონის დაბლა.    
  
  
ჩაწერილი მაკროების მოდიფიცირება   
VBA–ს საფუძვლების შესწავლის ერთ–ერთი კარგი საშუალებაა მაკროს შექმნა ჩამწერით და მერე მისი კოდის 
გაანალიზება. ხშირად ჩაწერილი მაკროები აუცილებელზე (საჭიროზე) მეტ კოდს შეიცავენ. ქვემოთ განვიხილოთ 
მაგალითი რომელიც ჩაწერილი მაკროს კოდის გაუმჯობესების შესაძლებლობას გვიჩვენებს.   
სავარჯიშო 2: ჩაწერილი მაკროს გაუმჯობესება  
   
1. ავირჩიოთ ნებისმიერი უჯრედი ან უჯრედთა ბლოკი.   
2. გავუშვათ მაკროს ჩამწერი და გამოვიძახოთ FormatCells (Home -> cells -> Format -> FormatCells). Use Relative 
References გამორთულია.   
3. FormatCells -> Font ბრძანებით ავირჩიოთ ფონტი „Times New Roman“ და წითელი ფერი – „Red“. FormatCells -> Fill –> 
Pattern Color ბრძანებით ავირჩიოთ ყვითელი ფერი, FormatCells -> Alignment ბრძანებით ავირჩიოთ Horizontal – 
„Center“,  FormatCells -> Number ბრძანებით ავირჩიოთ Currency.   
4. დავაწკაპუნოთ OK–ზე და გავაჩეროთ მაკროს ჩამწერი.   
5. დავათვალიეროთ კოდი VBA რედაქტორში:   
Sub Macro1()  
    Selection.NumberFormat = "$#,##0.00"  
    With Selection  
        .HorizontalAlignment = xlCenter  
        .VerticalAlignment = xlBottom  
        .WrapText = False  
        .Orientation = 0  
        .AddIndent = False  
        .IndentLevel = 0  
        .ShrinkToFit = False  
        .ReadingOrder = xlContext  
        .MergeCells = False  
    End With  
    With Selection.Font  
        .Name = "Times New Roman"  
        .FontStyle = "Ïîëóæèðíûé"  
        .Size = 14  
        .Strikethrough = False  
        .Superscript = False  
        .Subscript = False  
        .OutlineFont = False  
        .Shadow = False  
        .Underline = xlUnderlineStyleNone  
        .Color = 255  



        .TintAndShade = 0  
        .ThemeFont = xlThemeFontNone  
    End With  
    With Selection.Interior  
        .Pattern = xlNone  
        .TintAndShade = 0  
        .PatternTintAndShade = 0  
    End With  
End Sub  
  
6. ადვილი შესამჩნევია რომ კოდის გარკვეული ნაწილი ზესმეტია. წაშალეთ ზედმეტი სტრიქონები.   
7. დავრწმუნდეთ რომ კოდი  
  
       Sub Macro1  
            Selection.NumberFormat = "$#,##0.00"   
            With Selection   
                  .HorizontalAlignment = xlCenter  
            End With   
            With Selection.Font   
                 .Name = "Times New Roman"   
                 .ColorIndex = 3   
      End With  
      With Selection.Interior   
           .ColorIndex = 6   
      End With   
 End Sub   
იგივე სამუშაოს ასრულებს.  
  
8. დარწმუნდით რომ კოდი   
Sub Macro1  
  With Selection   
.NumberFormat = "$#,##0.00"   
.HorizontalAlignment = xlCenter  
.Font.Name = "TimesNewRoman"   
.Font.ColorIndex = 3   
.Interior.ColorIndex = 6   
   End With   
End Sub   
კვლავ იგივე სამუშაოს ასრულებს.  
  



ცვლადები 

ცვლადის გამოცხადება ხდება შემდეგი წესების დაცვით: 

■ ცვლადის სახელი იწყება ასოთი. 

■ ცვლადის სახელი არ უნდა შეიცავდეს 255 სიმბოლოზე მეტს. 

■ ცვლადის სახელი არ შეიძლება იყოს VBA ენის საკვანძო სიტყვა (მაგ. Dim ან Sub ან End). 

■ ცვლადის სახელი არ უნდა შეიცავდეს არცერთ ჰარს და არცერთ სიმბოლოს შემდეგი 

ჩამონათვლიდან: . ! # $ % & @. 

 

მთელი ტიპის ცვლადის გამოცხადება ხდება შემდეგნაირად: 

Dim a As Integer 

ზოგადად: 

Dim variableName As DataType 

 

 

VBA-ში გამოიყენება მონაცემთა შემდეგი ტიპები: 

■ String—სტრიქონი: 

Dim newFileName As String 

newFileName = “Budget Notes.doc” 

■ Date—თარიღი: 

Dim myBirthDate As Date 

Dim myBirthTime As Date 

Dim anotherDate As Date 

myBirthDate = #8/23/59# 

myBirthTime = #3:02 AM# 

anotherDate = #4/27/07 16:05# 

■ Object—ობიექტის ტიპის. 



■ Byte—ბაიტი, მთელი რიცხვი 0-დან 255-მდე. 

■ Integer—მთელი რიცხვი  –32,768-დან 32,767-მდე. 

Dim paragraphNumber As Integer 

paragraphNumber = 1 

■ Long—დიდი მთელი რიცხვები  –2,147,483,648-დან 2,147,483,647-მდე. 

Dim wordCount As Long 

wordCount = 100000 

■ Boolean—ბულის ტიპი, ღებულობს მნიშვნელობებს True ან False. 

Dim documentSaved As Boolean 

documentSaved = False 

■ Currency—თანხის ტიპი (ფული), –922,337,203,685,477.5808-დან 922,337,203,685,477.5807-მდე. 

■ Single—ნამდვილი ტიპის რიცხვი: 

Dim averageUnitSales As Single 

averageUnitSales = 50.3 

■ Double—ორმაგი სიზუსტის ნამდვილი რიცხვი: 

Dim atomsInTheUniverse As Double 

atomsInTheUniverse = 2.0E+79 

■ თუ ცვლადის გამოცხადებისას არ არის მითითებული მისი ტიპი, მაშინ მას ენიჭება ტიპი 

Variant. 

 

 

მონაცემთა ტიპის შეცვლა 

მოდულის ფარგლებში შესაძლებელია მონაცემთა ტიპის შეცვლა ბრძანებით DefType.  

მაგალითად, ბრძანება 

DefType letter1[-letter2] 

იმ ცვლადებს რომელთა სახელები იწყებიან letter1-ით (ან იწყებიან ასოთი letter1-სა და letter2 

შორის) შეუცვლის ტიპს, ამ ცვლადებს მიენიჭებათ ტიპი Type. 



მაგალითად: 

DefInt X-Z 

 

DefType ბრძანების საკვანძო სიტყვები. 

DefType      მონაცემთა ტიპი 

DefBool       Boolean 

DefByte       Byte 

DefInt         Integer 

DefLng       Long 

DefCur       Currency 

DefSng      Single 

DefDbl      Double 

DefDate    Date 

DefStr       String 

DefObj       Object 

DefVar       Variant 

 

მომხმარებლის მიერ შექმნილი მონაცემთა ტიპი 

 გარკვეულ შემთხვევებში მიზანშეწონილია შემოვიღოთ მონაცემთა ახალი ტიპი. მაგალითად, 

დავუშვათ რომ საქმე გვაქვს პროგრამასთან, რომელშიც განიხილება მანქანის მწარმოებელი 

ფირმა, მოდელი, გამოშვების წელი და ფასი. ამ ინფორმაციის დასამუშავებლად შეიძლება 

გამოგვეცხადა ცვლადები  

Dim carMake As String  

Dim carModel As String  

Dim yearMade As Integer  

Dim carPrice As Currency  

მაგრამ, თუ მანქანების რაოდენობა დიდია, აშკარაა ასეთი მიდგომის არაეფექტურობა. 

განვიხილოთ შემდეგი კოდი  



Type CarInfo  

make As String  

model As String  

made As Integer  

price As Currency  

End Type  

რომლითაც შემოღებულია მონაცემთა ახალი ტიპი CarInfo. მაშინ ყოველი ახალი მანქანისათვის 

გამოვაცხადებთ ცვლადს myCar  

Dim myCar As CarInfo  

ხოლო წვდომას თითოეულ ველზე განვახორციელებთ შემდეგნაირად  

myCar.make = “Porsche”  

myCar.model = “911 Turbo”  

myCar.made = 2007  

myCar.price = 122000 

 

მასივები 

ოცელემენტიანი მასივის სახელად employees გამოცხადება ხდება შემდეგნაირად: 

Dim employees(19) As String 

ელემენტზე წვდომა ხდება ინდექსის გამოყენებით, მაგალითად პირველ ელემენტს მივანიჭოთ 

სტრიქონი “Kartvelishvili”: 

employees(0) = “Kartvelishvili” 

თუ გვსურს რომ ელემენტთა ინდექსირება იწყებოდეს 1-ით, საჭიროა მოდულის თავში 

დავამატოთ ბრძანება 

Option Base 1 

 

 

 



დინამიური მასივები 

დინამიური მასივის გამოცხადება ხდება შემდეგნაირად: 

Dim myArray() As Double 

თუ პროგრამის მსვლელობაში გაირკვა ელემენტთა საჭირო რაოდენობა, მაგალითად 123, მაშინ 

მოვახდენთ myArray() მასივის ხელახალ გამოცხადებას ReDim ბრძანების გამოყენებით: 

ReDim myArray(123) 

 

განვიხილოთ მაგალითი რომელშიც Excel წიგნის ფურცელთა სახელები შეინახებიან დინამიურ 

მასივში: 

Sub StoreWorksheetNames() 

Dim sheetNames() As String 

Dim totalSheets As Integer 

Dim sheet As Worksheet 

Dim i As Integer 

Dim strMessage As String 

‘ 

‘ Store the total number of worksheets 

‘ that are in the current workbook 

‘ 

totalSheets = ActiveWorkbook.Worksheets.Count 

‘ 

‘ Now redimension the dynamic array 

‘ 

ReDim sheetNames(totalSheets) 

‘ 

‘ Loop through the worksheets to store the names in the array 

‘ 



For i = 1 To totalSheets 

sheetNames(i - 1) = ActiveWorkbook.Worksheets(i).Name 

Next ‘i 

‘ 

‘ Loop through the array to add the names to a string 

‘ 

strMessage = “Here are the worksheet names:” & vbCrLf 

For i = 0 To totalSheets - 1 

strMessage = strMessage & sheetNames(i) & vbCrLf 

Next ‘i 

‘ 

‘ Display the worksheet names 

‘ 

MsgBox strMessage 

End Sub 

 

მრავალგანზომილებიანი მასივები 

ორგანზომილებიანი (20x2 განზომილების) მასივის გამოცხადება: 

Dim employees(19,1) As String 

ელემენტზე წვდომა: 

employees(0,0) = “Giorgi” 

employees(0,1) = “Kartvelishvili” 



მომხმარებლის მიერ შემოღებული მონაცემთა ტიპები 
 
გარკვეულ შემთხვევებში მიზანშეწონილია შემოვიღოთ მონაცემთა ახალი ტიპი. 
მაგალითად, დავუშვათ რომ საქმე გვაქვს პროგრამასთან, რომელშიც განიხილება 
მანქანის მწარმოებელი ფირმა, მოდელი, გამოშვების წელი და ფასი. ამ ინფორმაციის 
დასამუშავებლად შეიძლება გამოგვეცხადა ცვლადები  
 

Dim carMake As String  
Dim carModel As String  
Dim yearMade As Integer  
Dim carPrice As Currency  
 
მაგრამ, თუ მანქანების რაოდენობა დიდია, აშკარაა ასეთი მიდგომის არაეფექტურობა. 
განვიხილოთ შემდეგი კოდი 
 
Type CarInfo  
  make As String  
  model As String  
  made As Integer  
  price As Currency  
End Type 
 

რომლითაც შემოღებულია მონაცემთა ახალი ტიპი CarInfo. მაშინ ყოველი ახალი 
მანქანისათვის გამოვაცხადებთ ცვლადს myCar 
Dim myCar As CarInfo 

ხოლო წვდომას თითოეულ ველზე განვახორციელებთ შემდეგნაირად  
myCar.make = “Porsche”  
myCar.model = “911 Turbo”  
myCar.made = 2007  
myCar.price = 122000  
 

 

ფუნქცია  MsgBox 
 

ფუნქციის სინტაქსი ასეთია:  
MsgBox(Prompt[,Buttons][,Title][,HelpFile][,Context]) 

 

 

Prompt     შეტყობინება რომელიც გამოვა დიალოგურ ფანჯარაში 

Buttons    რიცხვი, რომლის მიხედვითაც მოხდება ფანჯრის ღილაკების შერჩევა 
Title        ფანჯრის სათაური 



HelpFile     შეტყობინება დახმარების გამოყენების შესახებ 
Context      რიცხვი, რომელიც მოახდენს დახმარების მომსახურე ფაილის       
                       იდენტიფიცირებას 
 

ჩვეულებრივ MsgBox დიალოგურ ფანჯარას აქვს მხოლოდ OK ღილაკი. სხვა ღილაკების 
გამოყენება შესაძლებელია Buttons პაერამეტრის შერჩევით. ქვემოთ მოყვანილ 
ცხრილში მოცემულია ყველა შესაძლო ოფცია 

კონსტანტა მნიშვნელობა აღწერა 
vbOKOnly 0 მხოლოდ OK ღილაკი (default)  
vbOKCancel 1 OK და Cancel ღილაკები 
vbAbortRetryIgnore 2 Abort, Retry და Ignore ღილაკები 
vbYesNoCancel 3 Yes, No და Cancel ღილაკები 
vbYesNo 4 Yes და No ღილაკები 
vbRetryCancel 5 Retry და Cancel ღილაკები 
იარლიყი   
vbCritical 16 იარლიყი  Critical  
vbQuestion 32 იარლიყი  Question 
vbExclamation 48 იარლიყი  Exclamation 
vbInformation 64 იარლიყი  Information 
ფოკუსი ღილაკზე   
vbDefaultButton1 0 გააქტიურებულია პირველი ღილაკი 
vbDefaultButton2 256 გააქტიურებულია მეორე ღილაკი 
vbDefaultButton3 512 გააქტიურებულია მესამე ღილაკი 
მოდალობა   
vbApplicationModal 0 მიმდინარე აპლიკაცია ვერ გააგრძელებს 

მუშაობას  MsgBox –ის ღილაკზე დაჭერის 
გარეშე 

vbSystemModal 4096 ვერცერთი აპლიკაცია ვერ გააგრძელებს 
მუშაობას  MsgBox –ის ღილაკზე დაჭერის 
გარეშე 

 

Buttons პარამეტრის შერჩევა ხდება არჩეული ოფციების შესატყვისი კონსტანტების 
შეკრების გზით. ასევე შესაძლებელია კონსტანტების ჩამოთვლა მათ შორის „+“ ნიშნის 
გამოყენებით, მაგალითად:  vbYesNo + vbQuestion + vbDefaultButton2. 

განვიხილოთ მაგალითი: 

Sub ButtonTest() 
Dim msgPrompt As String, msgTitle As String 
Dim msgButtons As Integer, msgResult As Integer 



msgPrompt = “Are you sure you want to display “ & vbCrLf & _ 
“the worksheet names?” 
msgButtons = vbYesNo + vbQuestion + vbDefaultButton2 
msgTitle = “Display Worksheet Names” 
msgResult = MsgBox(msgPrompt, msgButtons, msgTitle) 
End Sub 
 

MsgBox ფუნქციის დაბრუნებული მნიშვნელობები მოიცემა შემდეგი ცხრილით 

კონსტანტა მნიშვნელობა 
vbOK 1 
vbCancel 2 
vbAbort 3 
vbRetry 4 
vbIgnore 5 
vbYes 6 
vbNo 7 

 

განვიხილოთ მაგალითი: 

Sub ButtonTest2() 
Dim msgPrompt As String, msgTitle As String 
Dim msgButtons As Integer, msgResult As Integer 
msgPrompt = “Are you sure you want to display “ & vbCrLf & _ 
“the worksheet names?” 
msgButtons = vbYesNo + vbQuestion + vbDefaultButton2 
msgTitle = “Display Worksheet Names” 
msgResult = MsgBox(msgPrompt, msgButtons, msgTitle) 
If msgResult = vbYes Then 
StoreWorksheetNames 
End If 
End Sub 

 

ფუნქცია InputBox 
 

ფუნქციის სინტაქსი ასეთია:  
InputBox(Prompt[,Title][,Default][,Xpos][,Ypos][,HelpFile][,Context]) 

 

Prompt     შეტყობინება რომელიც გამოვა დიალოგურ ფანჯარაში 

Title        ფანჯრის სათაური 
Default   მნიშვნელობა გაჩუმებით 



 Xpos, Ypos დიალოგური ფანჯრის მდებარეობა  

 HelpFile შეტყობინება დახმარების გამოყენების შესახებ  

 Context რიცხვი, რომელიც მოახდენს დახმარების მომსახურე ფაილის იდენტიფიცირებას  

 

განვიხილოთ მაგალითი:  

 

Sub InputBoxTest()  

Dim inputData As String  

‘ 

‘ მონაცემის აღება  

‘ 

inputData = InputBox(“Enter some text:”, “Input Box Text”)  

‘ 

‘ გადამოწმება, მიღებულია თუ არა რაიმე მონაცემი  

‘ 

If inputData <> “” Then  

‘ 

‘ თუ მონაცემი მიღებულია, მაშინ მისი გამოყვანა MsgBox ფუნქციის გამოყენებით  

‘ 

MsgBox inputData  

End If  

End Sub 



VBA: 2. განმეორების ოპერატორები 

(VBA02-Loops.xls) 
გახსენით დოკუმენტი VBA02-Loops.xls და დაათვალიერეთ ოთხი ფურცელი. თითოეული 
შეიცავს რიცხვთა ორ სვეტს (სვეტი A და სვეტი B). გვესაჭიროება მაკროს გამოყენებით 
გამოვიანგარიშოთ რიცხვთა საშუალო არითმეტიკული თითოეული სტრიქონისათვის.  

გახსენით Visual Basic Editor–ი (Alt+F11) და შეხედეთ კოდს Module1– ში.  თქვენ 
დაინახავთ სხვადასხვა მაკროებს. ქვემოთმოყვანილ სავარჯიშოებში, თავდაპირველად გაუშვით 
მაკროები აღსრულებაზე და შემდეგ შეხედეთ კოდს, გაანალიზეთ კოდი. 

 
სავარჯიშო 1: Do… Loop Until… 
მაკროს მიზანია გაირბინოს სვეტი C და შეიტანოს საჭირო მნიშვნელობები საჭირო უჯრედებში. 

ფურცელზე Sheet1 აირჩიეთ უჯრედი C2 და გაუშვით აღსრულებაზე მაკრო Loop1. 

შესაბამისი კოდი: 

Sub Loop1() 
' This loop runs until there is nothing in the next column 
    Do 
    ActiveCell.FormulaR1C1 = "=Average(RC[-1],RC[-2])" 
    ActiveCell.Offset(1, 0).Select 
    Loop Until IsEmpty(ActiveCell.Offset(0, 1)) 
End Sub 

მაკრო გამოითვლის მარცხენა ორ უჯრედში ჩაწერილი მნიშვნელობების საშუალო 
არითმეტიკულს,  ჩაწერს აქტიურ უჯრედში C2, და შემდეგ გააქტიურდება მის ქვემოთ მყოფი 
უჯრედი C3, შემდეგ კვლავ გაიმეორებს იგივეს C სვეტის უჯრედებისათვის ვიდრე მის 
მარჯვნივ მყოფი D სვეტის უჯრედი არ არის ცარიელი. წაშალეთ მონაცემები უჯრედებში 
C2:C20 და გადადით შემდეგ სავარჯიშოზე. 

 
სავარჯიშო 2: Do While… Loop 
 

ფურცელზე Sheet1 აირჩიეთ უჯრედი C2 და გაუშვით აღსრულებაზე მაკრო Loop2. 

შესაბამისი კოდი: 

Sub Loop2() 
' This loop runs as long as there is something in the next column 
    Do While IsEmpty(ActiveCell.Offset(0, 1)) = False 
    ActiveCell.FormulaR1C1 = "=Average(RC[-1],RC[-2])" 
    ActiveCell.Offset(1, 0).Select 
    Loop 
End Sub 



ეს მაკრო შეასრულებს იგივეს, ოღონდ განსხვავებული გზით. გაითვალისწინეთ, რომ IsEmpty 
= False ნიშნავს "არ არის ცარიელი". წაშალეთ მონაცემები უჯრედებში C2:C20 და გადადით 
შემდეგ სავარჯიშოზე. 

 

სავარჯიშო 3: Do While Not… Loop 
 

ფურცელზე Sheet1 აირჩიეთ უჯრედი C2 და გაუშვით აღსრულებაზე მაკრო Loop3. 

შესაბამისი კოდი: 

Sub Loop3() 
' This loop runs as long as there is something in the next column 
    Do While Not IsEmpty(ActiveCell.Offset(0, 1)) 
    ActiveCell.FormulaR1C1 = "=Average(RC[-1],RC[-2])" 
    ActiveCell.Offset(1, 0).Select 
    Loop 
End Sub 

ეს მაკრო შეასრულებს იგივეს. წაშალეთ მონაცემები უჯრედებში C2:C20 და გადადით შემდეგ 
სავარჯიშოზე. 

 

სავარჯიშო  4: IF ოპერატორი 
 

გადადით Sheet2 ფურცელზე,  აირჩიეთ უჯრედი C2 და გაუშვით აღსრულებაზე მაკრო 
Loop4. 

შესაბამისი კოდი: 

Sub Loop4() 
' This loop runs as long as there is something in the next column 
' It does not calculate an average if there is already something in 
the cell 
    Do 
    If IsEmpty(ActiveCell) Then 
        ActiveCell.FormulaR1C1 = "=Average(RC[-1],RC[-2])" 
    End If 
    ActiveCell.Offset(1, 0).Select 
    Loop Until IsEmpty(ActiveCell.Offset(0, 1)) 
End Sub 

 

 

 

სავარჯიშო  5: 

გახსენით ახალი წიგნი Exell-ში, გადადით VBA რედაქტორში, შექმენით პროცედურა რომლის 
კოდი მოყვანილია ქვემოთ. გაუშვით პროცედურა აღსრულებაზე და გაანალიზეთ პროცედურის 
კოდი. 



Sub myoffset() 
Dim reply As String 
reply = "yes" 
ActiveSheet.Range(“A1”).Select 
Do While reply = "yes" 
ActiveCell = Application.InputBox("Enter a name: ", "MyName", , , , , 2) 
ActiveCell.Offset(0, 1). Select 
ActiveCell = Application.InputBox("Enter the salary: ", "MySalary", , , , , 1) 
ActiveCell.Offset(0, 1). Select 
ActiveCell = Application.InputBox("Enter Perks: ", "MyPerks", , , , , 1) 
ActiveCell.Offset(1, -2). Select 
reply = InputBox("Do you want to continue(yes/no)") 
If reply = "no" Then End 
Loop 
End Sub 
 



მომხმარებლის მიერ შემოღებული მონაცემთა ტიპები 
 
გარკვეულ შემთხვევებში მიზანშეწონილია შემოვიღოთ მონაცემთა ახალი ტიპი. 
მაგალითად, დავუშვათ რომ საქმე გვაქვს პროგრამასთან, რომელშიც განიხილება 
მანქანის მწარმოებელი ფირმა, მოდელი, გამოშვების წელი და ფასი. ამ ინფორმაციის 
დასამუშავებლად შეიძლება გამოგვეცხადა ცვლადები  
 

Dim carMake As String  
Dim carModel As String  
Dim yearMade As Integer  
Dim carPrice As Currency  
 
მაგრამ, თუ მანქანების რაოდენობა დიდია, აშკარაა ასეთი მიდგომის არაეფექტურობა. 
განვიხილოთ შემდეგი კოდი 
 
Type CarInfo  
  make As String  
  model As String  
  made As Integer  
  price As Currency  
End Type 
 

რომლითაც შემოღებულია მონაცემთა ახალი ტიპი CarInfo. მაშინ ყოველი ახალი 
მანქანისათვის გამოვაცხადებთ ცვლადს myCar 
Dim myCar As CarInfo 

ხოლო წვდომას თითოეულ ველზე განვახორციელებთ შემდეგნაირად  
myCar.make = “Porsche”  
myCar.model = “911 Turbo”  
myCar.made = 2007  
myCar.price = 122000  
 

 

ფუნქცია  MsgBox 
 

ფუნქციის სინტაქსი ასეთია:  
MsgBox(Prompt[,Buttons][,Title][,HelpFile][,Context]) 

 

 

Prompt     შეტყობინება რომელიც გამოვა დიალოგურ ფანჯარაში 

Buttons    რიცხვი, რომლის მიხედვითაც მოხდება ფანჯრის ღილაკების შერჩევა 
Title        ფანჯრის სათაური 



HelpFile     შეტყობინება დახმარების გამოყენების შესახებ 
Context      რიცხვი, რომელიც მოახდენს დახმარების მომსახურე ფაილის       
                       იდენტიფიცირებას 
 

ჩვეულებრივ MsgBox დიალოგურ ფანჯარას აქვს მხოლოდ OK ღილაკი. სხვა ღილაკების 
გამოყენება შესაძლებელია Buttons პაერამეტრის შერჩევით. ქვემოთ მოყვანილ 
ცხრილში მოცემულია ყველა შესაძლო ოფცია 

კონსტანტა მნიშვნელობა აღწერა 
vbOKOnly 0 მხოლოდ OK ღილაკი (default)  
vbOKCancel 1 OK და Cancel ღილაკები 
vbAbortRetryIgnore 2 Abort, Retry და Ignore ღილაკები 
vbYesNoCancel 3 Yes, No და Cancel ღილაკები 
vbYesNo 4 Yes და No ღილაკები 
vbRetryCancel 5 Retry და Cancel ღილაკები 
იარლიყი   
vbCritical 16 იარლიყი  Critical  
vbQuestion 32 იარლიყი  Question 
vbExclamation 48 იარლიყი  Exclamation 
vbInformation 64 იარლიყი  Information 
ფოკუსი ღილაკზე   
vbDefaultButton1 0 გააქტიურებულია პირველი ღილაკი 
vbDefaultButton2 256 გააქტიურებულია მეორე ღილაკი 
vbDefaultButton3 512 გააქტიურებულია მესამე ღილაკი 
მოდალობა   
vbApplicationModal 0 მიმდინარე აპლიკაცია ვერ გააგრძელებს 

მუშაობას  MsgBox –ის ღილაკზე დაჭერის 
გარეშე 

vbSystemModal 4096 ვერცერთი აპლიკაცია ვერ გააგრძელებს 
მუშაობას  MsgBox –ის ღილაკზე დაჭერის 
გარეშე 

 

Buttons პარამეტრის შერჩევა ხდება არჩეული ოფციების შესატყვისი კონსტანტების 
შეკრების გზით. ასევე შესაძლებელია კონსტანტების ჩამოთვლა მათ შორის „+“ ნიშნის 
გამოყენებით, მაგალითად:  vbYesNo + vbQuestion + vbDefaultButton2. 

განვიხილოთ მაგალითი: 

Sub ButtonTest() 
Dim msgPrompt As String, msgTitle As String 
Dim msgButtons As Integer, msgResult As Integer 



msgPrompt = “Are you sure you want to display “ & vbCrLf & _ 
“the worksheet names?” 
msgButtons = vbYesNo + vbQuestion + vbDefaultButton2 
msgTitle = “Display Worksheet Names” 
msgResult = MsgBox(msgPrompt, msgButtons, msgTitle) 
End Sub 
 

MsgBox ფუნქციის დაბრუნებული მნიშვნელობები მოიცემა შემდეგი ცხრილით 

კონსტანტა მნიშვნელობა 
vbOK 1 
vbCancel 2 
vbAbort 3 
vbRetry 4 
vbIgnore 5 
vbYes 6 
vbNo 7 

 

განვიხილოთ მაგალითი: 

Sub ButtonTest2() 
Dim msgPrompt As String, msgTitle As String 
Dim msgButtons As Integer, msgResult As Integer 
msgPrompt = “Are you sure you want to display “ & vbCrLf & _ 
“the worksheet names?” 
msgButtons = vbYesNo + vbQuestion + vbDefaultButton2 
msgTitle = “Display Worksheet Names” 
msgResult = MsgBox(msgPrompt, msgButtons, msgTitle) 
If msgResult = vbYes Then 
StoreWorksheetNames 
End If 
End Sub 

 

ფუნქცია InputBox 
 

ფუნქციის სინტაქსი ასეთია:  
InputBox(Prompt[,Title][,Default][,Xpos][,Ypos][,HelpFile][,Context]) 

 

Prompt     შეტყობინება რომელიც გამოვა დიალოგურ ფანჯარაში 

Title        ფანჯრის სათაური 
Default   მნიშვნელობა გაჩუმებით 



Xpos, Ypos  დიალოგური ფანჯრის მდებარეობა 
HelpFile     შეტყობინება დახმარების გამოყენების შესახებ 
Context      რიცხვი, რომელიც მოახდენს დახმარების მომსახურე ფაილის       
                       იდენტიფიცირებას 
 

განვიხილოთ მაგალითი: 

Sub InputBoxTest() 
Dim inputData As String 
‘ 
‘ მონაცემის აღება 
‘ 
inputData = InputBox(“Enter some text:”, “Input Box Text”) 
‘ 
‘ გადამოწმება, მიღებულია თუ არა რაიმე მონაცემი 
‘ 
If inputData <> “” Then 
‘ 
‘ თუ მონაცემი მიღებულია, მაშინ მისი გამოყვანა  MsgBox ფუნქციის გამოყენებით  
‘ 
MsgBox inputData 
End If 
End Sub 

 

არითმეტიკული ოპერატორები 

ოპერატორი სახელი მაგალითი  შედეგი 
+ შეკრება 10+5 15 

– გამოკლება 10-5 5 
– მოპირდპირე –10 –10 

* გამრავლება 10*5 50 

/ გაყოფა 10/5 2 

\ მთელი გაყოფა 11\2 5 

^ ახარისხება 2^5 32 

Mod ნაშთით გაყოფა 10 Mod 3 1 

 

 

 

 



კონკატენაციის ოპერატორი & 

კონკატენაციის ოპერატორით ხდება ორი სტრიქონის გაერთიანება. მაგალითად: 

Dim strFirst As String 
Dim strLast As String 
strFirst = “Gia” 
strLast = “Giorgashvili” 
MsgBox strFirst & “ “ & strLast 

პროცედურა  MsgBox–ით გამოიყვანს  

Gia Giorgashvili 

 

შედარების ოპერატორები 

ოპერატორი სახელი მაგალითი  შედეგი 
= ტოლია 10=5 False 
> მეტია 10>5 True 
< ნაკლებია 10<12 True 

>= მეტია ან ტოლი 10>=5 True 

<= ნაკლებია ან ტოლი 10<=5 False 
<> არ უდრის 11<>2 True 

 

ლოგიკური ოპერატორები 

ოპერატორი ზოგადი სახე დაბრუნებული მნიშვნელობა 
And Expr1 And Expr2 True თუ ორივე გამოსახულება 

ჭეშმარიტია, წინააღმდეგ 
შემთხვევაში False 

Or Expr1 Or Expr2 True თუ ამ 
გამოსახულებებიდან ერთი 
მაინც ჭეშმარიტია, წინააღმდეგ 
შემთხვევაში False 

Xor Expr1 Xor Expr2 False თუ გამოსახულებების 
ლოგიკური მნიშვნელობები 
ტოლია, წინააღმდეგ 
შემთხვევაში True 

Not Not Expr True თუ გამოსახულება 
მცდარია, False თუ 
გამოსახულება ჭეშმარიტია 



 

 

ოპერატორთა შესრულების პრიორიტეტები 

  

ოპერატორი მოქმედება პრიორიტეტი 
^ ახარისხება პირველი 
– მოპირდაპირე მეორე 
*      / გამრავლება, გაყოფა მესამე 
\ მთელი გაყოფა მეოთხე 
Mod ნაშთით გაყოფა მეხუთე 
+     - შეკრება, გამოკლება მეექვსე 
& კონკატენაცია მეშვიდე 
=    <    >    <=   >=    <> შედარება მერვე 
And Or Xor Not ლოგიკური მეცხრე 
 

ფრჩხილების გამოყენებით შესაძლებელია გამოსახულებაში ოპერატორთა შესრულების 
პრიორიტეტების მართვა, მაგალითად: 

3^(15/(5*2–5))= 3^(15/(10–5))= 3^(15/5)= 3^3 = 27 

 

 

ზოგიერთი სასარგებლო ფუნქცია 

შემთხვევითი რიცხვის გენერირების ფუნქცია Rnd()  ქმნის ფსევდო–შემთხვევით რიცხვს 
[0, 1] შუალედში. ნამდვილად შემთხვევითი რიცხვის მიღებისათვის Rnd()  ფუნქციის 
წინ უნდა გამოვიყენოთ ბრძანება Randomize, მაგალითად: 

Randomize 
myRandomNumber = Rnd() 

ქვემოთ მოყვანილი კოდი ქმნის შემთხვევით მთელ რიცხვს [a, b] დიაპაზონიდან 

Int((b - a) * Rnd() + a)  

 

 



განვიხილოთ სტრიქონებთან მომუშავე ზოგიერთი ფუნქცია.  

ფუნქცია  Asc(string) დაგვიბრუნებს string სტრიქონის პირველი სიმბოლოს ANSI 
კოდს.  

ფუნქცია InStr(start,string1,string2) დაგვიბრუნებს იმ სიმბოლოს პოზიციას, 
საიდანაც იწყება სტრიქონი string2 სტრიქონში string1, დაწყებული start–იდან. 

Mid$(string,start,length) დაგვიბრუნებს string სტრიქონის length 
რაოდენობის სიმბოლოს, დაწყებული start–იდან. 

ფუნქცია Mid(string,start,length) დააბრუნებს იგივეს, ოღონდ როგორც 
Variant ტიპისა, და არა  String. 

Left$(string,length) დააბრუნებს length რაოდენობის სიმბოლოს string 
სტრიქონიდან, დაწყებული მარცხნიდან. 

Len(string) დააბრუნებს string სტრიქონის სიმბოლოთა რაოდენობას. 

 

განვიხილოთ მაგალითი: 

Function ExtractLastName(fullName As String) As String 
Dim spacePos As Integer 
spacePos = InStr(fullName, “ “) 
ExtractLastName = Mid$(fullName, spacePos + 1,_ 
Len(fullName) - spacePos) 
End Function 

Sub TestIt() 
MsgBox ExtractLastName(“Giorgi Giorgashvili”) 
End Sub 

 

 

თარიღის ტიპის (Date) მონაცემებთან მუშაობისას უნდა გვახსოვდეს რომ VBA–ში ყოველ 
კონკრეტულ თარიღს შეესაბამება ე.წ. სერიული რიცხვი. დეკემბერი 31, 1899 თარიღი 
გამოიყენება როგორც ათვლის წერტილი, მაგალითად,  იანვარი 1, 1900 თარიღის 
სერიული რიცხვი არის 1, იანვარი 2, 1900–ის სერიული რიცხვია 2. დროისათვის VBA 
ენაში გამოიყენება აგრეთვე სერიული რიცხვები 0–დან 1–დე, ათვლა იწყება შუაღამეს 
24:00 – მისი სერიული რიცხვია. ქვემოთ მოყვანილ ცხრილში მოცემულია სერიული 
რიცხვები დროის ზოგიერთი მნიშვნელობისათვის. 



  

სერიული რიცხვი დრო 
0.25 6:00:00 AM 
0.375 9:00:00 AM 
0.70833 5:00:00 PM 
.99999 11:59:59 PM 
 

მთელი და წილადი სერიული რიცხვების შეკრებით მივიღებთ თარიღს დროის 
ჩათვლით, მაგალითად  39317.5  წარმოადგენს  აგვისტო 23, 2007, 12:00. 

ფუნქცია CDate(expression) მოახდენს expression გამოსახულების 
კონვერტირებას  სერიულ რიცხვში. Day(date) დაგვიბრუნებს date თარიღის დღეებს. 

ფუნქცია Now დაგვიბრუნებს მიმდინარე თარიღს და დროს. Year(date) 
დაგვიბრუნებს date თარიღის წელს. 

განვიხილოთ მაგალითი, რომელშიც პიროვნების დაბადების თარიღის მიხედვით 
გამოითვლება მისი ასაკი. 

Function CalculateAge(birthDate As Date) As Byte 
Dim birthdayNotPassed As Boolean 
birthdayNotPassed = CDate(Month(birthDate) & “/” & _ 
Day(birthDate) & “/” & _ 
Year(Now)) > Now 
CalculateAge = Year(Now) - Year(birthDate) + birthdayNotPassed 
End Function 
‘ 
‘ CalculateAge პროცედურის შემოწმება 
‘ 
Sub TestIt2() 
MsgBox CalculateAge(#8/23/59#) 
End Sub 



მართვის ოპერატორები 

 

If ... Then 

თუ პირობა condition  ჭეშმარიტია, მაშინ სრულდება ინსტრუქცია statement: 

If condition Then statement 
 
თუ ინსტრუქციათა რაოდენობა ერთზე მეტია, მაშინ: 
 
If condition Then 
[statements] 
End If 

 

If...Then...Else 

თუ პირობა condition  ჭეშმარიტია, მაშინ სრულდება ინსტრუქციები 
TrueStatements, წინააღმდეგ შემთხვევაში FalseStatements: 

If condition Then 
[TrueStatements] 
Else 
[FalseStatements] 
End If 

 

 

Select Case 

Select Case TestExpression 
Case FirstCaseList 
[FirstStatements] 
Case SecondCaseList 
[SecondStatements] 
<etc> 
Case Else 
[ElseStatements] 
End Select 

TestExpression – გამოსახულება, რომელიც გვიბრუნებს კონკრეტულ მნიშვნელობას 
(რიცხვითი, სტრიქონული,...), CaseList – TestExpression გამოსახულების შესაძლო 
მნიშვნელობები, Statements – ინსტრუქციები. თუ TestExpression გამოსახულების 



მნიშვნელობას შეესაბამება ერთზე მეტი მნიშვნელობა  CaseList–ში, მაშინ სრულდება 
პირველი შემთხვევის ინსტრუქციები. 

 

მაგალითად, განვიხილოთ ფუნქცია LetterGrade(rawScore As Integer), 
რომელიც დაგვიბრუნებს შეფასებას შემდეგი ცხრილის შესაბამისად 

Raw Score                        Letter Grade 
90 და ზევით                             A 
80 – 89                                        B 
70 – 79                                         C 
60 – 69                                        D 
60 და ნაკლები                           F 

 

Function LetterGrade(rawScore As Integer) As String 
Select Case rawScore 
Case Is < 0 
LetterGrade = “ERROR! Score less than 0!” 
Case Is < 60 
LetterGrade = “F” 
Case Is < 70 
LetterGrade = “D” 
Case Is < 80 
LetterGrade = “C” 
Case Is < 90 
LetterGrade = “B” 
Case Is <= 100 
LetterGrade = “A” 
Case Else 
LetterGrade = “ERROR! Score greater than 100!” 
End Select 
End Function 
 

 

ფუნქცია IIf 

IIf (condition, TrueResult, FalseResult) 

თუ პირობა condition ჭეშმარიტია, დაგვიბრუნებს TrueResult, თუ არა, მაშინ 
დაგვიბრუნებს FalseResult. 

 



ფუნქცია Choose 

Choose(index, value1, value2,...) 

ფუნქცია დააბრუნებს რიგით მე–index  ცვლადის მნიშვნელობას. აქ index  მთელი 
დადებითი რიცხვია.  

 

 ფუნქცია Switch 

Switch(expr1, value1, expr2, value2,...) 

ფუნქცია დააბრუნებს პირველივე expri  ჭეშმარიტი გამოსახულების შემდეგ მყოფ 
მნიშვნელობას valuei. 

მაგალითად ჩვენს მიერ განხილული ფუნქცია LetterGrade(rawScore As 
Integer)შეიძლება ჩაიწეროს ასე: 

Function LetterGrade(rawScore As Integer) As String 
LetterGrade2 = Switch( _ 
rawScore < 0, “ERROR! Score less than 0!”, _ 
rawScore < 60, “F”, _ 
rawScore < 70, “D”, _ 
rawScore < 80, “C”, _ 
rawScore < 90, “B”, _ 
rawScore <= 100, “A”, _ 
rawScore > 100, “ERROR! Score greater than 100!”) 
End Function 

განმეორების ოპერატორები 

 

კოდები განხილულია Microsoft Excel–ის გარემოსათვის. 

  

იტერაცია ობიექტთა კოლექციაში (For Each…Next)  

პროგრამის ობიექტური მოდელი გულისხმობს ობიექტთა კოლექციებს. მაგალითად, Excel–ის 
აპლიკაციის ობიექტი (Application object) შეიცავს წიგნთა კოლექციას. წიგნი შეიცავს 
ფურცელთა ერთობლიობას, ფურცელი შეიცავს უჯრედთა, სტრიქონთა და სვეტთა 
კოლექციებს, და ა.შ. 

 

VBA–ში ხშირად იყენებენ იტერაციას For Each…Next, რათა მოვახდინოთ იტერირება 
კოლექციის თითოეულ წევრზე. კოდი იწყება განსახილველ ობიექტთა ტიპის ცვლადის 
გამოცხადებით. ქვემოთ მოყვანილ მაგალითში ხდება მოცემული წიგნის ფურცლების 
გადამოწმება იმ მიზნით არსებობს თუ არა მათ შორის ისეთი, რომლის სახელია NewSheet. თუ 
არ არსებობს, მაშინ ის ქმნის ახალ ფურცელს სახელით NewSheet.  

Sub AddSheet_1() 
Dim objSheet As Worksheet  
For Each objSheet In ActiveWorkbook.Sheets  

If objSheet.Name = "NewSheet" Then Exit Sub  
Next objSheet 
Worksheets.Add  
ActiveSheet.Name = "NewSheet"  

End Sub  

ოპერატორი For Each…Next იწვევს კოდის განმეორებას იმდენჯერ, რამდენი ობიექტიცაა 
კოლექციაში.  

If ოპერატორი დაასრულებს პროცედურას Exit Sub ბრძანებით იმ შემთხვევაში, თუ 
ფურცლებს შორის არსებობს ისეთი, რომლის სახელია NewSheet. წინააღმდეგ შემთხვევაში 
Worksheets.Add ბრძანება დაამატებს ახალ ფურცელს, ხოლო ActiveSheet.Name = 
"NewSheet"  ბრძანება მისცემს საჭირო სახელს. 

 

განვიხილოთ განხილული მაგალითის მოდიფიცირებული ვარიანტი: 

Sub AddSheet_2() 
Dim objSheet As Worksheet  
For Each objSheet In ActiveWorkbook.Sheets  

If objSheet.Name = "NewSheet" Then 
MsgBox "A worksheet with that name already exists." 
Worksheets("NewSheet").Activate 
Exit Sub  

End If  
Next objSheet 
Worksheets.Add  
ActiveSheet.Name = "NewSheet"  

End Sub  

შემდეგი მაგალითი უჯრედთა ერთობლიობაში (კოლექციაში) ეძებს ისეთს, რომელიც შეიცავს 
წინასწარ მოცემულ მნიშვნელობას, ამ შემთხვევაში დღევანდელ თარიღს. თუ ასეთი უჯრედი 
მოიძებნა, მაშინ ვღებულობთ შეტყობინებას, საჭირო უჯრედი მოინიშნება და პროცედურა 
დასრულდება. თუ საჭირო უჯრედი არ არსებობს, მაშინ ვღებულობთ შესაბამის შეტყობინებას  

Sub FindCell_1 () 
Dim objCell As Object 
For Each objCell In ActiveSheet.UsedRange.Cells 

If objCell.Value = Date Then 
 objCell.Select 
MsgBox "Today's date was found in cell: " & objCell.Address 
Exit Sub  

End If  
Next objCell 
MsgBox "No cell containing today's date was 

found." 

End Sub  

კოდში UsedRange არის გააქტიურებული ფურცლის გამოყენებული უჯრედების 
ერთობლიობა – დიაპაზონი. ფაქტიურად ძებნა ხდება ამ დიაპაზონში, თუმცა ჩვენ 
შეგვიძლია შევცვალოთ ძებნის დიაპაზონი, მაგალითად ასე:  

For Each objCell In ActiveSheet.Range("A1:C25").Cells  

ან მაგალითად რომელიმე კონკრეტულ სვეტში (მაგალითად მეორე სვეტში):  

For Each objCell In ActiveSheet.Columns(2).Cells  

ან მაგალითად რაიმე წინასწარ მონიშნულ დიაპაზონში:  

For Each objCell In Selection.Cells  

 

ამ მაგალითში, როგორც კი თარიღის შემცველი უჯრედი მოძებნილია, ბრძანებით Exit Sub 
ხდება პროცედურის დასრულება, მაგრამ, იქნებ კიდევ არსებობს სხვა უჯრედი რომელშიც ასევე 
თარიღია ჩაწერილი? ბუნებრივია ძიება რომ გაგრძელდეს უნდა ამოვიღოთ ბრძანება Exit Sub, 
მაგრამ ამას შეიძლება მოჰყვეს გაუთვალისწინებელი შედეგები. საჭიროა შემოვიღოთ რაიმე 
სტრიქონის ტიპის ცვლადი, სადაც თარიღის შემცველი უჯრედების მისამართები ჩაიწერება. 
იტერაციის დასრულების შემდეგ თუ შემოღებული ცვლადი ცარიელია, ე.ი. მასში სიმბოლოთა 
რაოდენობა უდრის 0, მაშინ ასეთი უჯრედი არ არსებობს, ხოლო თუ ასეთი უჯრედები 
არსებობენ, მაშინ ხდება მძიმეების მოშორება სტრიქონულ ცვლადში და საჭირო უჯრედების 
მონიშვნა. 
  
Sub FindCell_2() 

Dim objCell As Object  
Dim strRange As String  
For Each objCell In ActiveSheet.UsedRange.Cells 

If objCell.Value = Date Then 
 strRange = strRange & "," & objCell.Address 

End If  
Next objCell 
If Len(strRange) = 0 Then 

 MsgBox "No cell containing today's date was found." 
Else strRange = Right(strRange, Len(strRange) - 1) 

Range(strRange).Select 
MsgBox "Today's date appears in the selected cells" 

     End If 
End Sub  
 

 

განმეორების ოპერატორიდან გამოსვლა მაკროს გაგრძელებით  
ასეთ შემთხვევაში გამოვიყენებთ Exit For ბრძანებას ნაცვლად Exit Sub ბრძანებისა. Exit 
For ბრძანება იწვევს გადასვლას Next–ის მომდევნო ბრძანებაზე. 
  
Sub AddSheet_3() 
     Dim objSheet As Worksheet  
     Dim blnFound As Boolean  
     For Each objSheet In ActiveWorkbook.Sheets  

If objSheet.Name = "NewSheet" Then  
    blnFound = True  
    Exit For  
End If  
Next objSheet 
If Not blnFound Then 
 Worksheets.Add  
 ActiveSheet.Name = "NewSheet"  

End If 
 MsgBox "This workbook contains " & Worksheets.Count & " sheets." 

End Sub  

აქ ბრძანება Exit For წყვეტს ოპერატორს თუ მოიძებნა საჭირო ფურცელი და ბულის ტიპის 
ცვლადს ანიჭებს მნიშვნელობას True. If ოპერატორი ამოწმებს ბულის ტიპის ცვლადის 
მნიშვნელობას და თუ ის არ არის False, ქმნის ახალ ფურცელს. მაკრო გრძელდება და გამოაქვს 
შეტყობინება წიგნში ფურცელთა რაოდენობაზე.  

ოპერატორი For…Next 

თუ ჩვენ ვიცით რაიმე ბრძანების გამეორების რაოდენობა, მაშინ მიზანშეწონილია გამოვიყენოთ 
ოპერატორი For…Next. ამისათვის, როგორც წესი, საჭიროა მთვლელს მიენიჭოს საწყისი 
მნიშვნელობა (ჩვეულებრივ 1 ან 0) და ბოლო მნიშვნელობა:  

For i = 1 To 25  

ან კიდევ, გარკვეული გზით გამოთვლილ მნიშვნელობამდე, რომელიც რაიმე ცვლადშია 
ჩაწერილი:  

For i = 1 To Selection.Rows.Count  

 

ქვემოთ მოყვანილ მაგალითში ხდება "A" სვეტში ჩაწერილი რიცხვების შემოწმება 
კენტ–ლუწობაზე და შედეგი იწერება "B" სვეტში. სვეტში მოძრაობისას ხდება თითოეული 
უჯრედის მონიშვნა.  

Sub OddOrEven_1() 
     Dim intRowCount As Integer  
     Dim i As Integer 
     intRowCount = Range("A1").CurrentRegion.Rows.Count 
     Range("A1").Select 
     For i = 1 To intRowCount  

If ActiveCell.Value / 2 = Int(ActiveCell.Value / 2) Then  
     ActiveCell.Offset(0, 1).Value = "Even" 

Else  
     ActiveCell.Offset(0, 1).Value = "Odd" 
End If 

 ActiveCell.Offset(1, 0).Select 

Next i  

End Sub  

კოდი გამართულია, თუმცა არსებობს უფრო ეფექტური კოდიც, სადაც მთვლელი "i" 
გამოიყენება როგორც  Offset – ის კომპონენტი: 
  
Sub OddOrEven_2() 
      Dim intRowCount As Integer 
      Dim i As Integer 
      intRowCount = Range("A1").CurrentRegion.Rows.Count 
      For i = 0 To intRowCount - 1  

If Range("A1").Offset(i, 0).Value / 2 = Int(Range("A1").Offset(i, 
0).Value / 2) Then  

    Range("A1").Offset(i, 1).Value = "Even" 
Else  
    Range("A1").Offset(i, 1).Value = "Odd" 

   End If                             
      Next i  
End Sub  
 

გარდა ამისა, მთვლელისათვის შეიძლება ცვლილების ბიჯის (Step) გამოყენება:  

For i = 1 To 100 Step 2  

ციკლის შესრულება შეიძლება შებრუნებული რიგით, როცა ბიჯი უარყოფითია:  

For i = 100 To 1 Step -1  

 

განმეორების ოპერატორი Do…Loop Until 

ოპერატორი გამოიყენება მაშინ, როცა წინასწარ არ ვიცით ბრძანებების განმეორების 
რაოდენობა. ოპერატორი წყვეტს მოქმედებას როცა შესრულდება გარკვეული პირობა.  
ქვემოთ მოყვანილ მაგალითში ხდება სვეტში ცარიელი უჯრის ძებნა.  

Sub FindEmptyCell_1() 
      Range("A1").Select 
Do 
 ActiveCell.Offset(1, 0).Select 
Loop Until IsEmpty(ActiveCell.Value) 
 
End Sub 
 

 

Do…Loop Until ოპერატორის სინტაქსი ასე გამოიყურება:  

Do 'Code goes here 
Loop Until <Condition>  

... რომელიც პირველ უჯრედს არ შეეხება, ხოლო:  

Do Until <Condition>  

'Code goes here 

Loop  

... პირველ უჯრედს შეეხება.  

  

 
 
 
 
 

განმეორების ოპერატორი Do While…Loop 

ოპერატორის სინტაქსი ასეთია:  

Do While <Condition>  
'Code goes here 

Loop  

ან:  

Do 'Code goes here 
Loop While <Condition>  

განვიხილოთ მაგალითი: 

Sub FindEmptyCell_3() 
Range("A1").Select 
Do While Not IsEmpty(ActiveCell.Value) 

ActiveCell.Offset(1,0).Select 
Loop 

End Sub  



ობიექტები 

VBA ენაში ობიექტი, ისევე როგორც სხვა ობიექტზე ორიენტირებულ დაპროგრამების 
ენებში, წარმოადგენს კლასის კონკრეტულ რეალიზაციას.    

ობიექტებთან ურთიერქმედება VBA-ში შესაძლებელია სამი გზით: 

• ობიექტის თვისებების შეცვლით 
• ობიექტის მეთოდების გამოყენებით 
• გარკვეული პროცედურის ინიციირებით ობიექტის რაიმე ხდომილების 

საპასუხოდ  

 

განვიხილოთ მანქანა როგორც ობიექტი. რა შეიძლება გვესმოდეს „თვისება, მეთოდი და 
ხდომილება“ ტერმინების ქვეშ? ბუნებრივი იქნებოდა შემდეგი პასუხი: 

„თვისება“ – მისი ფერი, ძრავის სიმძლავრე, მისი სიჩქარე, და ა.შ. 

„მეთოდი“ – აჩქარება, დამუხრუჭება, მობრუნება, და ა.შ. 

„ხდომილება“ – პარკირებული მანქანის ხელით შეხებისას ირთვება სიგნალიზაცია. 

 

თითოეული საოფისე აპლიკაცია ორგანიზებულია ობიექტთა იერარქიის გზით. 
ყველაზე ზოგადი ობიექტი არის თვითონ ეს აპლიკაცია, ხოლო დანარჩენ ობიექტებს 
გააჩნიათ საკუთარი ადგილი ამ იერარქიაში. მაგალითად, Excel–ის  წიგნის ფურცლები 
(Worksheets) იერარქიულად წიგნებს (Workbooks) მოსდევენ, ხოლო უჯრედები 
(Cells) კი ფურცლებს. თუ გვინდა მივმართოთ რაიმე ობიექტს ამ იერარქიაში, საჭიროა 
დავიწყოთ ზოგადი ობიექტით და მივუწეროთ თანმიმდევრობით ქვედა ობიექტები, 
გამოყოფილი ერთმანეთისაგან წერტილით. მაგალითად, თუ გვინდა მივმართოთ 
Memo.doc დოკუმენტის მეორე პარაგრაფის პირველ სიტყვას დავწერთ: 

Application.Documents(“Memo.doc”).Paragraphs(2).Range.Words(1) 

 

 

 



ობიექტთა თვისებები 

ყოველ ობიექტს გააჩნია განმსაზღვრელ მახასიათებელთა სიმრავლე. ამ მახასიათებლებს 
ობიექტის თვისებები ეწოდებათ. მაგალითად, Word–ის დოკუმენტის ობიექტს გააჩნია 
თვისება Name, აგრეთვე თვისება Save, რომელიც მიუთითებს არის თუ არა დოკუმენტი 
შენახული უკანასკნელი შენახვის მომენტიდან, და ა.შ. 

ობიექტის თვისებაზე მიმართვას აქვს შემდეგი სინტაქსი: 

Object.Property 

თუ გვინდა თვისებას მივანიჭოთ გარკვეული მნიშვნელობა, მაშინ დავწერთ: 

Object.Property=value 

აქ  value არის გამოსახულება, რომელიც გვაძლევს თვისებისადმი მისანიჭებელ 
მნიშვნელობას. მაგალითად: 

ActiveCell.Font.Size=12 
ActiveCell.Font.Name = ”Times New Roman” 
ActiveCell.Font.Italic=True 
 

ხშირად საჭიროა რაიმე თვისების მნიშვნელობის დადგენა, მაშინ დავწერთ 

variable=Object.Property 

მაგალითად: 

cellContents=ActiveCell.Value 

კოდი გააქტიურებული უჯრედის ჩანაწერს მიანიჭებს cellContents ცვლადს. 

 



შევნიშნოთ, რომ იერარქიის ქვედა დონის ობიექტები თავის მხრივ წარმოადგენენ ზედა 
დონის ობიექტთა თვისებებს, რაც ილუსტრირებულია ქვემოთ მოყვანილ სურათში.

 

 

ობიექტთა მეთოდები 

ობიექტის თვისებები აღწერენ მას, ხოლო ობიექტის მეთოდები აღწერენ მასზე 
შესაძლებელ მოქმედებებს. მიმართვა ობიექტის მეთოდზე დამოკიდებულია იმაზე, აქვს 
თუ არა მას არგუმენტები. თუ მეთოდს არგუმენტი არ გააჩნია, მაშინ მისი სინტაქსი 
ასეთია: 

Object.Method 

მაგალითად, შემდეგი ინსტრუქცია შეინახავს გააქტიურებულ დოკუმენტს 

ActiveDocument.Save 

თუ მეთოდს გააჩნია არგუმენტები, მაშინ მისი სინტაქსი ასეთია: 

Object.Method(argument1,argument2,…) 



მაგალითად, Word–ის დოკუმენტის ობიექტის მეთოდს  Close გააჩნია არგუმენტები, 
მისი სინტაქსი ასეთია: 

Object.Close(SaveChanges, OriginalFormat, RouteDocument) 

Object                            ობიექტი  Document, რომელთანაც ვმუშაობთ. 
SaveChanges         კონსტანტა, რომელიც მიანიშნებს იყო თუ არა ფაილი შენახული   
                                 დახურვამდე. 
OriginalFormat  კონსტანტა, რომელიც მიანიშნებს იყოს თუ არ ფაილი შენახული   
                                  საწყის ფორმატში. 
RouteDocument  True ან False, მიანიშნებს უნდა იყოს თუ არა დოკუმენტი   
                                  გადაგზავნილი სხვასთან. 
 

მაგალითად: 

ActiveDocument.Close wdPromptToSaveChanges,_ 
wdOriginalDocumentFormat,True 

ან: 

ActiveDocument.Close SaveChanges:=wdPromptToSaveChanges, _ 
OriginalFormat:=wdOriginalDocumentFormat, _ 
RouteDocument:=True 

მეორე შემთხვევაში მეთოდის არგუმენტების ჩამონათვალში გამოვიყენეთ ე.წ. 
დასახელებული არგუმენტები, რაც აუმჯობესებს პროცედურის კითხვადობას. გარდა 
ამისა ამ შემთხვევაში არაა აუცილებელი არგუმენტთა რიგითობის დაცვა. 

 

ჩამონათვალი WdOriginalFormat 

სახელი მნიშვნელობა აღწერა 

wdOriginalDocumentFormat 1 Original document format. 

wdPromptUser 2 Prompt user to select a document format. 

wdWordDocument 0 Microsoft Word document format. 

 

 

 



ობიექტთა ხდომილებების დამუშავება 

ხდომილება არის რაიმე მოვლენა ობიექტთან მიმართებაში. მაგალითად, დოკუმენტის 
გახსნა წარმოადგენს ხდომილებას ამ დოკუმენტის მიმართ. VBA–ენაში  მნიშვნელოვანი 
ადგილი უჭირავთ ხდომილების დამმუშავებელ პროცედურებს (event handlers), 
რომლებიც ობიექტის ხდომილების საპასუხოდ გაიშვებიან. მაგალითად, დოკუმენტის 
გახსნას, როგორც ხდომილებას, შეიძლება დავუკავშიროთ რაიმე პროცედურა, გახსნის 
ხდომილების დამმუშავებელი პროცედურა.  თუ VBA რედაქტორის Project ფანჯარაში 
მოვნიშნავთ ThisDocument, თაგვის მარჯვენა დაწკაპუნებით გამოყვანილ კონტექსტურ 
მენიუში ავირჩევთ View Code, და მარჯვნივ გამოსულ ფანჯარაში ობიექტთა სიიდან  

 

ამოვირჩევთ Document, ხოლო პროცედურების სიიდან კი Open, მივიღებთ კოდს 

Private Sub Document_Open() 

End Sub  

თუ მასში ჩავწერთ 

Private Sub Document_Open() 

MsgBox “Hello!” 

End Sub  

მაშინ Word-ის დოკუმენტის ყოველი გახსნისას  MsgBox ფუნქცია გამოიტანს შესაბამის 
მისალმებას. 

 

ობიექტთა კოლექცია 

ობიექტთა კოლექცია ეწოდება ერთი და იმავე ტიპის ობიექტთა სიმრავლეს. კოლექციის 
წევრს კოლექციის ელემენტი ეწოდება. კოლექციის ელემენტზე მიმართვა შესაძლებელია 
ან ობიექტის სახელით ან მისი ინდექსით. მაგალითად ბრძანება 



Documents(“Budget.doc”).Close 

დახურავს დოკუმენტს სახელით Budget.doc. თუ ეს დოკუმენტი გახსნილი იყო 
როგორც მესამე დოკუმენტი (ანუ მისი ინდექსია 3), მაშინ იგივეს გააკეთებდა ბრძანება 

Documents(3).Close 

 

როგორც ვიცით შესაძლებელია ობიექტის ტიპის ცვლადის გამოცხადება 

Dim variableName As ObjectType 

თუ ვმუშაობთ Document ტიპის ობიექტებთან, შეგვიძლია გამოვაცხადოთ ცვლადი 

Dim currentDocument As Document 

თუ გვინდა currentDocument ცვლადს მივანიჭოთ რაიმე კონკრეტული ობიექტი, 
მაგალითად დოკუმენტი Budget.doc, მაშინ დავწერთ 

Set currentDocument = Documents(“Budget.doc”) 

 

ობიექტთა შედარებისათვის არსებობს ოპერატორი Is, მაგალითად  

result = Object1 Is Object2 

ბრძანების თანახმად, თუ Object1 და Object2  ობიექტები დაემთხვა ერთმანეთს, 
მაშინ  result  მიიღებს მნიშვნელობას True, ხოლო თუ არ დაემთხვა, მაშინ 
მნიშვნელობას False. 

 

რადგან ობიექტებს გააჩნიათ ბევრი თვისება და მეთოდი, ხშირად ერთ ობიექტზე 
საჭიროა ჩატარდეს მრავალი მოქმედება, მაგალითად, როგორც ეს არის ნაჩვენები 
მიმდინარე დოკუმენტის (ThisDocument) პირველი პარაგრაფისათვის 
(Paragraphs(1)) 

Sub FormatParagraph() 
ThisDocument.Paragraphs(1).Style = “Heading 1” 
ThisDocument.Paragraphs(1).Alignment = wdAlignParagraphCenter 
ThisDocument.Paragraphs(1).Range.Font.Size = 16 
ThisDocument.Paragraphs(1).Range.Font.Bold = True 
ThisDocument.Paragraphs(1).Range.Font.Color = RGB(255, 0, 0) ‘ Red 
ThisDocument.Paragraphs(1).Range.Font.Name = “Times New Roman” 
End Sub 



კოდის შემოკლების მიზნით გამოიყენება ბრძანება  With, რომლის სინტაქსი ასეთია: 

With object 
  [statements] 
End With 

პროცედურა FormatParagraph() შეიძლება გადაიწეროს შემდეგნაირად: 

Sub FormatParagraph2() 
  With ThisDocument.Paragraphs(1) 
    .Style = “Heading 1” 
    .Alignment = wdAlignParagraphCenter 
    .Range.Font.Size = 16 
    .Range.Font.Bold = True 
    .Range.Font.Color = RGB(255, 0, 0) ‘ Red 
    .Range.Font.Name = “Times New Roman” 
  End With 
End Sub 

 

ობიექტი Application 

Application ობიექტი წარმოადგენს პროგრამის სხვა დანარჩენი ობიექტების ერთგვარ 
კონტეინერს.  განვიხილოთ მისი ზოგიერთი თვისება და მეთოდი.  

თვისება StatusBar გამოიყენება ფანჯრის სტატუსის პანელში რაიმე ტექსტური 
შეტყობინების გამოსაყვანად. მაგალითად: 

Sub StatusBarProperty() 
ActiveDocument.Save 
Application.StatusBar = ActiveDocument.Name & “ was saved.” 
End Sub 

პროცედურა დოკუმენტის შენახვის შემდეგ სტატუსის პანელში გამოიყვანს შესაბამის 
შეტყობინებას. 

ფანჯრის ტიტულის შეცვლის მიზნით გამოიყენება თვისება Caption, მაგალითად: 

Sub CaptionProperty() 
Application.Caption = “Hello Georgia” 
End Sub 

პროცედურა შეცვლის ფანჯრის ტიტულს. 

 

 



თვისება  WindowState –ს გააჩნია ცხრილში მოყვანილი სამი შესაძლო მნიშვნელობა 

WindowState Excel Word PowerPoint 
Maximized xlMaximized wdWindowStateMaximize ppWindowMaximized 
Minimized xlMinimized wdWindowStateMinimize ppWindowMinimized 
Normal xlNormal wdWindowStateNormal ppWindowNormal 
 

Application ობიექტს გააჩნია თვისება Dialogs, რომელიც წარმოადგენს ჩადგმულ 
დიალოგურ ფანჯარათა კოლექციას; თითოეული მათგანი არის ობიექტი Dialog. 
შევნიშნოთ რომ ისინი მხოლოდ Excel და Word პროგრამებში გამოიყენებიან.  

Word Constant Excel Constant Dialog Box 
wdDialogFormatFont xlDialogFont Font 
wdDialogFileNew xlDialogNew New 
wdDialogFileOpen xlDialogOpen Open 
wdDialogFilePageSetup xlDialogPageSetup Page Setup 
wdDialogEditPasteSpecial xlDialogPasteSpecial Paste Special 
wdDialogFilePrint xlDialogPrint Print 
wdDialogFilePrintSetup xlDialogPrinterSetup Printer Setup 
wdDialogFileSaveAs xlDialogSaveAs Save As 
wdDialogInsertObject xlDialogObject Object 
wdDialogFormatStyle xlDialogStyle Style 
wdDialogTableSort xlDialogSort Sort 
 

დიალოგური ფანჯრის გამოსაყვანად ვისარგებლებთ მეთოდით Show, მაგალითად 

Application.Dialogs(xlDialogPrint).Show 

 

თუ მომხმარებელი დააწკაპუნებს ღილაკზე Cancel, მაშინ მეთოდი Show დააბრუნებს 
მნიშვნელობას  False.  

If Not Application.Dialogs(xlDialogPrint).Show Then 
MsgBox “File was not printed” 
End If 

გარდა ამისა Show მეთოდს შეიძლება გააჩნდეს არგუმენტი 

Dialog.Show(Timeout) 
 
Dialog       ობიექტი   Dialog, რომელიც უნდა გამოჩნდეს 
Timeout      დრო, წამის მეათასადებში, რომლის გასვლის შემდეგ ფანჯარა გაქრება 

 



ობიექტი Window 

ობიექტი Window საერთოა ყველა აპლიკაციისათვის, ის აპლიკაციის მთავარი ფანჯარაა. 
ის წარმოადგენს აპლიკაციის (მაგალითად Word–ის) გახსნილი დოკუმენტების 
ფანჯრების კოლექციის ელემენტს. მასზე წვდომა შეიძლება განხორციელდეს ან 
ელემენტის რიგითი რიცხვით (მაგალითად ფანჯრის შესაბამისი დოკუმენტის ინდექსი 
3–ია, ანუ გახსნილია როგორც მესამე დოკუმენტი) ან დოკუმენტის სახელით,  ე.ი.: 

Windows(3) 
ან 
Windows(“Budget.doc”)    

 



Excel-ის დაპროგრამება 

იმისათვის რომ პროცედურაში გამოვიყენოთ Excel-ის რომელიმე ფუნქცია საჭიროა მის სახელს 
დაემატოს Application.WorksheetFunction. მაგალითად SUM() ფუნქციის გამოყენებისათვის 
დასახელებული Sales დიაპაზონისათვის უნდა დავწეროთ 

totalSales = Application.WorksheetFunction.Sum(Range(“Sales”))  

წიგნის ხელახალი გაანგარიშებისათვის გამოიყენება მეთოდები Calculate, რომელიც 
დაითვლის მხოლოდ იმ ფორმულებს რომლის უჯრედები შეიცვალა,  და 
Application.CalculateFull, რომელიც დაითვლის ყველა ფორმულას. 

მეთოდი Evaluate მოახდენს სტრიქონის კონვერტირებას ობიექტში: 

Evaluate(name), სადაც name უჯრედის მისამართი, დიაპაზონი ან დასახელებული 
დიაპაზონი. 

განვიხილოთ მაგალითი (შევიტანოთ კონკრეტული მნიშვნელობები წინასწარ): 

Sub EvaluateTest() 
Dim columnLetter As String 
Dim rowNumber As String 
Dim cellAddress As String 
‘ 
‘ Activate the “Sheet1” worksheet 
‘ 
Worksheets(“Sheet1”).Activate 
‘ 
‘ Get the value in cell A1 
‘ 
columnLetter = [A1].Value 
‘ 
‘ Get the value in cell B1 
‘ 
rowNumber = [B1].Value 
‘ 
‘ Concatenate the two values and then display the message 
‘ 
cellAddress = columnLetter & rowNumber 
MsgBox “The value in cell “ & cellAddress & “ is “ & Application.Evaluate(cellAddress) 
End Sub 

 

 

 



Application ობიექტს გააჩნია მეთოდი Wait, რომელიც შეაჩერებს პროცედურას გარკვეულ 
Time  დრომდე:  Application.Wait(Time). მაგალითად კოდი 

Application.Wait  Now + TimeValue(“00:00:05”) 

შეაჩერებს პროცედურას 5 წამით. 

 

თუ საჭიროა პროცედურის გაშვება რაიმე კონკრეტული დროისათვის გამოვიყენებთ 
OnTime  მეთოდს: 

Application.OnTime(EarliestTime,Procedure [,LatestTime] [, Schedule]) 

EarliestTime  პროცედურის გაშვების დრო 

Procedure     პროცედურის სახელი 

LatestTime    თუ პროგრამა EarliestTime დროისათვის არ არის მზად გაშვებისათვის, ის          

                      გაიმეორებს გაშვებას LatestTime დროისათვის. 

Schedule      True ან False. თუ გვსურს პროცედურის გაშვება ვირჩევთ   True  

 

აქ სასარგებლოა TimeValue(Time) ფუნქციის გამოყენება, სადაც არგუმენტი Time მიიღებს 
მნიშვნელობას სტრიქონის სახით, მაგალითად 

Application.OnTime   EarliestTime:=TimeValue(“5:00PM”), Procedure:=”Backup” 
 

Excel-ის კონკრეტულ წიგნზე წვდომისათვის გამოიყენება ან მისი სახელი, ან მისი 
ინდექსი (მისი გახსნის რიგი): 

Workbooks(“Budget.xlsx”) 

Workbooks(1) 

თუ გვაინტერესებს ის წიგნი, რომელშიც ჩვენი პროცედურა აღსრულდება, მაშინ 
გამოვიყენებთ ThisWorkbook. 

წიგნის გახსნისათვის გამოიყენება მეთოდი Open: 

Workbooks.Open(FileName),  სადაც FileName წარმოადგენს წიგნის სახელს თავისი 
ბილიკით ხისტ დისკზე, მაგალითად 



Workbooks.Open  „C:\NewFolder\Data.xlsx” 

ახალი წიგნის შექმნა ხდება Add მეთოდის გამოყენებით, მისი სინტაქსია 

Workbooks.Add(Template) 

არგუმენტი Template ოფციურია, თუ ის არ არის, მაშინ წიგნი შეიქმნება სტანდარტული 
წესით. 

განვიხილოთ პროცედურა რომელიც შემოგვთავაზებს ფურცელთა რაოდენობის არჩევას 
ახალი წიგნის შექმნის დროს. 

Sub NewWorkbookWithCustomSheets() 
Dim currentSheets As Integer 
With Application 
‘ 
‘ Save the current value of SheetsInNewWorkbook 
‘ 
currentSheets = .SheetsInNewWorkbook 
‘ 
‘ Ask how many sheets to include in the new workbook 
‘ and store the result in SheetsInNewWorkbook 
‘ 
.SheetsInNewWorkbook = InputBox( _ 
“How many sheets do you want “ & _ 
“in the new workbook?”, , 3) 
‘ 
‘ Create the new workbook 
‘ 
Workbooks.Add 
‘ 
‘ Restore the original value of SheetsInNewWorkbook 
‘ 
.SheetsInNewWorkbook = currentSheets 
End With 
End Sub 

 

განვიხილოთ აგრეთვე პროცედურა, რომელიც შეინახავს ყველა გახსნილ წიგნს. 

Sub SaveAll() 
Dim wb As Workbook 
Dim newFilename As Variant 
‘ 
‘ Run through all the open workbooks 
‘ 
For Each wb In Workbooks 



‘ 
‘ Has the workbook been saved before? 
‘ 
If wb.Path <> “” Then 
‘ 
‘ If so, save it 
‘ 
wb.Save 
Else 
‘ 
‘ If not, display the Save As dialog box 
‘ to get the workbook’s path & filename 
‘ 
With Application 
newFilename = .GetSaveAsFilename( _ 
FileFilter:=”Microsoft Office “ & _ 
“Excel Workbook “ & _ 
“(*.xlsx), *.xlsx”) 
End With 
‘ 
‘ Did the user click Cancel? 
‘ 
If newFilename <> False Then 
‘ 
‘ If not, save the workbook using the 
‘ specified path and filename 
‘ 
wb.SaveAs fileName:=newFilename 
End If 
End If 
Next ‘wb 
End Sub 

 

წიგნის დახურვა ხდება Close მეთოდის გამოყენებით, მისი სინტაქსია 

Workbook.Close([SaveChanges][, FileName][, RouteWorkbook]) 
Workbook                წიგნი, რომლის დახურვაც გვინდა 

SaveChanges         თუ წიგნში ცვლილებებია, მაშინ თუ SaveChanges  არის True - შენახვა   

                                             განხორციელდება, წინააღმდეგ შემთხვევაში -არა. თუ საერთოდ   
                            გამოტოვებულია, მაშინ მიმართავს მომხმარებელს, მოხდეს თუ არა   
                            შენახვა. 
FileName                    ფაილის სახელი  
RouteWorkbook    სხვა მიმღებთან გადაგზავნა (True, False). 
 
 



Excel-ის წიგნის კონკრეტულ ფურცელზე  წვდომისათვის გამოიყენება ან მისი სახელი, 
ან მისი ინდექსი (მისი რიგითი ნომერი წიგნში): 

Worksheets(“Sheet1”) 
Worksheets(1) 
 

 
ახალი ფურცლის დამატება 

 

Worksheets.Add([Before][, After][, Count][, Type]) 
 
Before      რომელი ფურცლის წინ დაემატოს. თუ Before და After გამოტოვებულია, მაშინ   
                 ფურცელი დაემატება გააქტიურებული ფურცლის წინ. 
After         რომელი ფურცლის შემდეგ დაემატოს.  
Count       დამატებულ ფურცელთა რაოდენობა. გამოტოვის შემთხვევაში დაემატება   
                 ერთი ფურცელი. 
Type        დამატებული ფურცლის ტიპი: xlWorksheet (the default), xlExcel4MacroSheet და 
                 xlExcel4IntlMacroSheet 
 
 
მაგალითად: 
 
Worksheets.Add Before:=Worksheets(“Sales”) 
 
 

Worksheet ობიექტის ზოგიერთი მეთოდი 
 

Worksheet.Activate  - ფურცლის გააქტიურება. 

Worksheet.Calculate - ფურცლის ფორმულების დათვლა. 

Worksheet.copy - ფურცლის კოპირება, მისი სრული სინტაქსია 

Worksheet.Copy([Before][, After]) 
 
Worksheet       ფურცელი, რომლის კოპირებაც გვსურს. 
Before             ფურცელი, რომლის წინ მოხდება კოპირება. გამოტოვების შემთხვევაში   
                        შეიქმნება ახალი წიგნი, სადაც მოხდება არჩეული ფურცლის კოპირება. 
After                ფურცელი, რომლის შემდეგ მოხდება კოპირება. 
 

მაგალითად კოდი 

Worksheets(“Budget 2007”).Copy 



მოახდენს კოპირებას ახალ წიგნში. 

Worksheet.Delete - ფურცლის წაშლა 

Worksheet.Move - ფურცლის გადაადგილება, მისი სინტაქსია 

Worksheet.Move([Before][, After]) 

Worksheet       ფურცელი, რომლის გადაადგილებაც გვსურს. 
Before             ფურცელი, რომლის წინ მოხდება გადაადგილება. გამოტოვების   
                        შემთხვევაში  შეიქმნება ახალი წიგნი, სადაც მოხდება არჩეული ფურცლის   
                        გადაადგილება. 
After                ფურცელი, რომლის შემდეგ მოხდება გადაადგილება. 
 

მაგალითად, 

Worksheets(“Budget 2007”).Move Before:=Worksheets(“Budget 2006”) 

 

Worksheet.Select – ფურცლის გააქტიურება 

 

Range  და  Cells  მეთოდები 

Worksheet.Range(Name) 
 
Worksheet    ობიექტი Worksheet, რომლის მეთოდსაც წარმოადგენს Range. გამოტოვების   
                     შემთხვევაში მეთოდი ეკუთვნის ობიექტს ActiveSheet. 
Name             დიაპაზონის სახელი ტექსტური სახით. 
 

მაგალითად,  კოდი8 

Worksheets(“Data”).Range(“Criteria”).ClearContents 

წაშლის “Criteria” სახელით დასახელებული დიაპაზონის ჩანაწერებს. 

 

დიაპაზონი 

Worksheet.Range(Cell1, Cell2) 
 



წარმოადგენს უჯრედთა მართკუთხა ფორმის ერთობლიობას, სადაც უჯრედები Cell1 და  
Cell2 წარმოადგენენ მარცხენა ზედა წვეროსა და მარჯვენა ქვედა წვეროს უჯრედებს. 
 

Worksheet.Cells(RowIndex, ColumnIndex)  წარმოადგენს უჯრედს რომელიც მდებარეობს  

RowIndex-ე სტრიქონისა და ColumnIndex-ე სვეტის გადაკვეთაზე. 

მაგალითად, კოდი 
 
For colNumber = 1 To 5 
Cells(1, colNumber).Value = “Field” & colNumber 
Next colNumber 

შეავსებს დიაპაზონს [A1:E1]  ჩანაწერებით Field1, Field2, Field3, Field4, Field5. 

 

თუ საჭიროა ვიმუშაოთ მთლიან სტრიქონთან ან სვეტთან უნდა გამოვიყენოთ 
მეთოდები Rows და Columns. 

Object.Rows([Index])  დააბრუნებს სტრიქონს: 
 
Object     ობიექტი Worksheet ან Range. გამოტოვების შემთხვევაში - ActiveSheet . 
Index      სტრიქონის ინდექსი. თუ ობიექტი worksheet-ია, ინდექსი 1 მიუთითებს პირველ   
               სტრიქონს. თუ ობიექტი არის range, მაშინ ინდექსი 1 მიუთითებს range   
               ობიექტის პირველ სტრიქონს. გამოტოვების შემთხვევაში მეთოდი დააბრუნებს   
               ობიექტის ყველა სტრიქონთა კოლექციას. 
 

განვიხილოთ მაგალითი სადაც ახალი სტრიქონი ჩაეკვრება ბოლო სტრიქონის წინ 
როგორიც არ უნდა იყოს გადაცემული დიაპაზონი rangeObject. 

Sub InsertRangeRow(rangeObject As Range) 
Dim totalRows As Integer, lastRow As Integer 
With rangeObject 
totalRows = .Rows.Count                ‘ Total rows in the range 
lastRow = .Rows(totalRows).Row   ‘ Last row number 
.Rows(lastRow).Insert                      ‘ Insert before last row 
End With 
End Sub 
 
Sub InsertTest() 
InsertRangeRow  ThisWorkbook.Worksheets(1).Range(“Test”) 
End Sub 



 

Object.Columns([Index])  დააბრუნებს სვეტს: 
 
Object     ობიექტი Worksheet ან Range. გამოტოვების შემთხვევაში - ActiveSheet . 
Index      სვეტის ინდექსი. თუ ობიექტი worksheet-ია, ინდექსი 1 მიუთითებს პირველ   
               სვეტს. თუ ობიექტი არის range, მაშინ ინდექსი 1 მიუთითებს range   
               ობიექტის პირველ სვეტს. გამოტოვების შემთხვევაში მეთოდი დააბრუნებს   
               ობიექტის ყველა სვეტთა კოლექციას. 
 
მაგალითად, კოდი 
 
Columns(2).ColumnWidth = 20 
 
გააქტიურებულ ფურცელში მეორე  სვეტს შეუცვლის სიგანეს.  



განვიხილოთ   Application ობიექტის ზოგიერთი თვისება და 
მეთოდი. 

 
 

შექმენით წიგნი Excel-ში და შეინახეთ. შემდეგ შეასრულეთ სავარჯიშო 

(ე.ი. წიგნი რომელშიც კოდი უნდა გაეშვას წინასწარ შენახული უნდა იყოს). 

ახალი წიგნი შეიქმნება იმავე დირექტორიაში სახელით  Test Scores.xls 

 
' ახალი წიგნის შექმნა 
Sub MakeNewWorkbook() 
Dim work_book As 
Workbook Dim 
work_sheet As 
Worksheet Dim i As 
Long 

 
' workbook –ის შექმნა 

  Set work_book = Workbooks.Add() 
 

' ზოგიერთი სვეტისათვის სახელების შექმნა  
Set work_sheet = work_book.Sheets(1)  
work_sheet.Name = "Test Scores"  
work_sheet.Cells(1, 1) = "Last Name"  
work_sheet.Cells(1, 2) = "First Name"  
For i = 1 To 10 

work_sheet.Cells(1, i + 2) = i 
Next i 

 
work_sheet.Range("A1:B1").ColumnWi
dth = 20 
work_sheet.Range("C1:L1").ColumnWi
dth = 5 

 
     work_book.SaveAs ThisWorkbook.Path & "\Test Scores.xls" 
 
 End Sub 
 
 

FindFile, GetOpenFilename, და  GetSaveAsFilename 
 
მეთოდი  FindFile გამოაჩენს  დიალოგურ  ფანჯარას  File  Open. თუ  
მომხმარებელი აირჩევს ფაილს, მეთოდი გახსნის ამ ფაილს და დააბრუნებს 
მნიშვნელობას True. თუ მომხმარებელი აირჩევს Cancel, მაშინ მეთოდი 
დააბრუნებს False. 

 
მოყვანილ  მაგალითში  მომხმარებელს  ეძლევა  საშუალება  მოძებნოს  წიგნი.  
თუ  ის აირჩევს რომელიმეს და დააწკაპუნებს ღილაკზე OK, მაშინ კოდი 
დაამატებს ახალ ფურცელს გააქტიურებულ წიგნში და მოახდენს ახლად 
გახსნილი წიგნის პირველი ფურცლის კოპირებას მასში, შემდეგ კი უკანასკნელი 
წიგნი დაიხურება. 

 
შექმენით წიგნი Excel-ში და შეინახეთ. შემდეგ შეასრულეთ სავარჯიშო 



(ე.ი. წიგნი რომელშიც კოდი უნდა გაეშვას წინასწარ შენახული უნდა იყოს). 

 
' წიგნის პირველი ფურცლის კოპირება 
Private Sub cmdCopyFirstWorksheet_Click() 
Dim dest_ws As Worksheet 
Dim source_ws As Worksheet 

If Application.FindFile() Then 
' ახალი ფურცლის შექმნა 
Set dest_ws = ThisWorkbook.Worksheets.Add(, _ 

ThisWorkbook.Worksheets(ThisWorkbook.Worksheets.Count)) 
 

' ახალი წიგნის პირველი ფურცელზე წვდომა 
Set source_ws = ActiveWorkbook.Sheets(1) 

 
' კოპირება 
source_ws.UsedRange.Copy dest_ws.Range("A1") 

 
' ახალი წიგნის დახურვა 
ActiveWorkbook.Close 

     End If 
End Sub 

 
მეთოდი GetOpenFilename გვიჩვენებს დიალოგურ ფანჯარას File Open.  
თუ მომხმარებელი აირჩევს ფაილს, მეთოდი დაგვიბრუნებს არჩეული ფაილის 
სახელს. 

 
თუ   MultiSelect პარამეტრის მნიშვნელობა არის True, მაშინ მომხმარებელს 
შეუძლია აირჩიოს ერთზე მეტი ფაილი. ამ შემთხვევაში მეთოდი დააბრუნებს 
ფაილების სახელთა მასივს. თუ მომხმარებელი არ გააკეთებს არჩევანს, მეთოდი 
დააბრუნებს False. 

 
შევნიშნოთ,  რომ  თუ  MultiSelect არის  True,  მაშინ  ერთი  ფაილის  
შემთხვევაშიც  კი GetOpenFilename დაგვიბრუნებს სახელს მასივის სახით. 
მოყვანილ მაგალითში ხდება ერთი ან მეტი ფაილის არჩევა და შემდგომ მათი 
ჩვენება 

 
' ფაილის/ების არჩევა 
  

Private Sub cmdSelectFiles_Click() 
Dim result As Variant 
Dim txt As String 
Dim i As Integer 
result = Application.GetOpenFilename(MultiSelect:=True) 
If VarType(result) = vbBoolean Then 
    MsgBox "Canceled" 
Else 
    txt = "Selected:" & vbCrLf 
    For i = LBound(result) To UBound(result) 
        txt = txt & " " & result(i) & vbCrLf 
    Next i 
    MsgBox txt 
End If 
End Sub 
 
 



მეთოდი GetSaveAsFilename გამოაჩენს დიალოგურ ფანჯარას Save As. 
GetOpenFilename მეთოდის მსგავსად ის დაგვიბრუნებს არჩეული ფაილის 
სახელს, და False–ს თუ არჩევანი არ არის გაკეთებული. 

 
 
 
გადაკვეთა და გაერთიანება 

 
ფუნქცია Intersect დააბრუნებს Range ტიპის ობიექტს, რომელიც წარმოადგენს 
ორი ან რამოდენიმე Range ტიპის ობიექტების თანაკვეთას. მაგალითად, 
შემდეგი კოდი ქმნის L- ფორმის Range ტიპის ობიექტებს, მათი თანაკვეთაა C5 და F2 
უჯრედები. 

 
Sub 
SelectIntersecti
on() Dim r1 As 
Range 
Dim r2 As Range 
Dim r3 As Range 

 
' პირველი ობიექტი იყოს ყვითელი  
Set r1 = Range("C2:C6,C2:G2")  
r1.Interior.Color = vbYellow 

 
' მეორე ობიექტი წითელია 
Set r2 = Range("B5:F5,F1:F5") 
r2.Interior.Color = vbRed 
' თანაკვეთა იყოს ფორთოხლისფერი 
Set r3 = Application.Intersect(r1, r2) 
r3.Interior.Color = RGB(255, 128, 0)  

End Sub 
 
თუ  Intersect ფუნქციის Range ტიპის პარამეტრები არ თანაიკვეთებიან, მაშინ 
ფუნქცია არაფერს არ დააბრუნებს. 

 
ფუნქცია Union დააბრუნებს ორი ან მეტი Range ობიექტების გაერთიანებას 
როგორც Range ობიექტს. განვიხილოთ მაგალითი: 
 
Sub SelectUnion()  
Dim r1 As Range  
Dim r2 As Range 

 
Set r1 = Range("C2:C6,C2:G2")  
Set r2 = Range("B5:F5,F1:F5") 
Application.Union(r1, r2).Select 

 
End Sub 

 
 
 
 
 
 
 



მეთოდი OnKey გამოიყენება „ცხელი კლავიშის“ დასანიშნად. 
 
' F8 ღილაკი გაუშვებს F8Pressed პროცედურას 
Sub InstallF8() 

Application.OnKey "{F8}", "ThisWorkbook.F8Pressed" 
End Sub 

 
Sub F8Pressed() 

MsgBox "F8 pressed" 
  End Sub 

 
‘ცხელი კლავიშის გაუქმება 
 
Sub UninstallF8()  
Application.OnKey "{F8}" 

  End Sub 

 
 
OnTime მეთოდი 

 
OnTime მეთოდი გაუშვებს პროცედურას მოგვიანებით, მეთოდის პირველი 
პარამეტრი აღნიშნავს გაშვების დროს, ხოლო მეორე – პროცედურის სახელს. 
ოფციური მესამე პარამეტრი  – პროცედურის გაშვების უკანასკნელ დროს, 
ოფციური მეოთხე პარამეტრი Schedule  –     აპირებთ  თუ  არა  პროცედურის  
გაშვებას,  თუ  აუქმებთ  მაშინ Schedule:=False 

 
შემდეგი პროცედურა დაგეგმავს პროცედურის გაშვებას 5, 10, და 15 წამის 
პერიოდით, თუმცა მერე ხდება 10 წამინი გაშვების გაუქმება. 

 
Sub ScheduleTimes()  
Dim t1 As Date 
Dim t2 As Date 
Dim t3 As Date 

 
t1 = Now + TimeValue("00:00:05")  
t2 = t1 + TimeValue("00:00:05")  
t3 = t2 + TimeValue("00:00:05") 

 
 

Application.OnTime t1, "DisplayTime"  
Application.OnTime t2, "DisplayTime"  
Application.OnTime t3, "DisplayTime" 

 
Application.OnTime t2, "DisplayTime",  
Schedule:=False 

End Sub 

Sub DisplayTime()  
 MsgBox Now 
End Sub 
 

 
 
 
 



Copy, Delete და Move 
 
მეთოდი  Copy გამოიყენება ფურცლის კოპირებისას, მას ორი ოფციური 
პარამეტრი აქვს Before და  After, რომლებიც ახალი ფურცლის პოზიციას 
განსაზღვრავენ. მაგალითად კოდი 

 
ActiveSheet.Copy 
After:=Sheets(Sheets.Count) 

მოახდენს  აქტიური  ფურცლის  კოპირებას  ბოლო  ფურცლის  შემდეგ  შექმნილ 
ფურცელში. თუ პარამეტრები Before და  After არ არიან, მაშინ კოპირება 
მოხდება ახალ წიგნში. 

 
მეთოდი  Delete წაშლის  ფურცელს  წიგნიდან.  წინასწარ  მომხმარებელი  
იქნება გაფრთხილებული შესაბამისი შეტყობინებით. თუ შეტყობინება საჭირო 
არ არის, მაშინ დავწერთ DisplayAlerts = False, 

 
Application.DisplayAlerts = False 
Sheets(Sheets.Count).Delete 
Application.DisplayAlerts = True 

 
მეთოდი  Move გადაიტანს  ფურცელს ახალ პოზიციაზე, მაგალითად 

 
ActiveSheet.Move 
After:=Sheets(Sheets.Count) 
ActiveSheet.Move Before:=Sheets(1) 

 
 
 
 
 
ციკლოიდის გრაფიკი 

 
Sub DrawCycloid()  
Const PI = 3.14159265 
Dim X As Single 
Dim Y As Single Dim T As Single Dim dT As Single 

 
' პირველი წერტილი 

T = 0 
dT = PI / 20 
X = Xt(T)  
Y = Yt(T) 
 

' FreeformBuilder  ობიექტის  შექმნა 
With ActiveSheet.Shapes.BuildFreeform(msoEditingAuto, X, Y) 
' დანარჩენი წერტილების დამატება 
T = T + dT 
Do While T < 14 * PI 
X = Xt(T)  
Y = Yt(T) 
.AddNodes msoSegmentCurve, msoEditingAuto, X, Y  
T = T + dT 
Loop 
' უკანასკნელი წერტილის დამატება 



X = Xt(T) 
Y = Yt(T) 
.AddNodes msoSegmentCurve, msoEditingAuto, X, Y 
 
' FreeformBuilder–ის გარდაქმნა Shape ტიპის ობიექტში 
  With .ConvertToShape 
    .Fill.Visible = msoFalse 
  End With  
             End With  
End Sub 
 
 
Function Xt(T As Single) As Single 

Xt = 150 * (27 * Cos(T) + 15 * Cos(T * 20 / 7)) / 42 + 180 
End Function 

 
Function Yt(T As Single) As Single 

Yt = 150 * (27 * Sin(T) + 15 * Sin(T * 20 / 7)) / 42 + 180 
End Function 
 
 
 



 
User Forms  
სასწავლო კურსის არჩევის ფორმა 
საჭიროა შეიქმნას ფორმა, რომელიც სასწავლო კურსის გავლის კანდიდატის მონაცემებს “OK” 
ღილაკზე დაჭერის შედეგად შეიტანს MS Excel დოკუმენტში.  

ეს ფორმა იყენებს ტექსტურ ველებს (textbox), 
კომბინირებულ ტექსტურ ველებს (combobox), 
ოფციურ ღილაკებს, თოლიებს (checkbox)  და 
ღილაკებს (commandbutton).  

როცა მომხმარებელი დააწკაპუნებს OK ღილაკზე, 
ფორმის მონაცემები შეიტანება MS Excel-ის 
ფურცლის მომდევნო სტრიქონში.  

ფორმის აღწერა:  

ორი ტექსტური ველი (Name: და Phone:) და ორი 
კომბინირებული ტექსტური ველი (Department და 
Course), რომლებიც მომხმარებელს არჩევანის 
გაკეთების საშუალებას მისცემენ.  

გვაქვს სამი ოფციური ღილაკი (Introduction, 
Intermediate and Advanced) ერთ ჩარჩოში, 
რომლებიც კურსის შესწავლის დონეს 
განსაზღვრავენ.  

გვაქვს ორი თოლია (Lunch Required და Vegetarian), 
აქედან მეორე არ არის გააქტიურებული, ვინაიდან თუ 

ლანჩი არ არის არჩეული, სულაც არ არის საჭირო იმის ცოდნა, არის თუ არა მომხმარებელი 
ვეგეტარიანელი.  

გვაქვს სამი ღილაკი (OK, Cancel და Clear Form).  

 

 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 



მართვის ელემენტების თვისებების მნიშვნელობები:  
 

 
 

ფორმის აგება: 
1. გახსენით Excel-ის დოკუმენტი და გამოიძახეთ Visual Basic Editor–ი.  

2. შემდეგ მენიუს პანელში აირჩიეთ Insert –> UserForm.  

3. თუ Toolbox-ი არ ჩანს, მაშინ მენიუს პანელში აირჩიეთ View –> Toolbox.  

4. ააგეთ სურათზე მოყვანილი ფორმა 
 

 

კოდის დამატება:  
ფორმის ინიციალიზება:  

მოვახდინოთ ფორმის ინიციალიზება:  

1. ფორმის კოდის გამოჩენისათვის ავირჩიოთ View –> Code ან F7.  
2. კოდის ფანჯრის გახსნისას UserForm_Click() პროცედურა ცარიელია. აირჩიეთ ზევით 
განთავსებულ კომბინირებულ ტექსტურ ველებში UserForm და Initialize. ამით შეიქმნება საჭირო 
პროცედურა. ახლა შეგვიძლია წავშალოთ UserForm_Click() პროცედურა.  

3. შევიტანოთ პროცედურაში შემდეგი კოდი:  
 
   Private Sub UserForm_Initialize() 
            txtName.Value = " " 
            txtPhone.Value = " " 
            With cboDepartment  
                 .AddItem "Sales"  

Control  Type  Property  Setting  
UserForm  UserForm  Name  frmCourseBooking  
  Caption  Course Booking Form  
Name  Text Box  Name  txtName  
Phone  Text Box  Name  txtPhone  
Department  Combo Box  Name  cboDepartment  
Course  Combo Box  Name  cboCourse  
Level  Frame  Name  fraLevel  
  Caption  Level  
Introduction  Option Button  Name  optIntroduction  
Intermediate  Option Button  Name  optIntermediate  
Advanced  Option Button  Name  optAdvanced  
Lunch Required  Check Box  Name  chkLunch  
Vegetarian  Check Box  Name  chkVegetarian  
  Enabled  False  

OK  Command 
Button  Name  cmdOk  

  Caption  OK  
  Default  True  

Cancel  Command 
Button  Name  cmdCancel  

  Caption  Cancel  
  Cancel  True  

Clear Form  Command 
Button  Name  cmdClearForm  



                 .AddItem "Marketing"  
                 .AddItem "Administration"  
                 .AddItem "Design"  
                 .AddItem "Advertising"  
                 .AddItem "Dispatch"  
                 .AddItem "Transportation"  
            End With  
            cboDepartment.Value = " "  
            With cboCourse  
                 .AddItem "Access"  
                 .AddItem "Excel"  
                 .AddItem "PowerPoint"  
                 .AddItem "Word"  
                 .AddItem "FrontPage"  

End With  
cboCourse.Value = " "  
optIntroduction = True  
chkLunch = False  
chkVegetarian = False  
txtName.SetFocus  

End Sub  
 

ღილაკების ამუშავება:  
ღილაკი Cancel:  

1. ფორმაზე განლაგებულ Cancel ღილაკზე მოვახდინოთ ორმაგი დაწკაპუნება. გაიხსნება 
ფორმის კოდის ფანჯარა და კურსოსრი მოთავსდება cmdCancel_Click() პროცედურის შიგნით.  

2. ჩავწეროთ კოდი:  
 

Private Sub cmdCancel_Click() 
Unload Me  
End Sub  

 

ღილაკი Clear Form:  

1 ფორმაზე განლაგებულ Clear Form ღილაკზე მოვახდინოთ ორმაგი დაწკაპუნება. გაიხსნება 
ფორმის კოდის ფანჯარა და კურსოსრი მოთავსდება cmdClearForm_Click() პროცედურის შიგნით. 

2 ჩავწეროთ კოდი:  
 
       Private Sub cmdClearForm_Click() 
       Call UserForm_Initialize  
       End Sub  
 

ღილაკი OK:  

კოდის ეს ნაწილი ასრულებს ფორმის მონაცემების შეტანას Excel–ის დოკუმენტის ფურცელში.  

1. ფორმაზე განლაგებულ OK ღილაკზე მოვახდინოთ ორმაგი დაწკაპუნება. გაიხსნება ფორმის 
კოდის ფანჯარა და კურსოსრი მოთავსდება cmdOK_Click() პროცედურის შიგნით.  

2. ჩავწეროთ კოდი:  



Private Sub cmdOK_Click() 
ActiveWorkbook.Sheets("Course Bookings").Activate 
Range("A1").Select 
Do  
  If IsEmpty(ActiveCell) = False Then  
    ActiveCell.Offset(1,0).Select 
  End If 
Loop Until IsEmpty(ActiveCell) = True  
ActiveCell.Value = txtName.Value  
ActiveCell.Offset(0,1) = txtPhone.Value 
ActiveCell.Offset(0,2) = cboDepartment.Value 
ActiveCell.Offset(0,3) = cboCourse.Value 
If optIntroduction = True Then  
   ActiveCell.Offset(0,4).Value = "Intro" 
ElseIf optIntermediate = True Then  
   ActiveCell.Offset(0,4).Value = "Intermed" 
Else 
 ActiveCell.Offset(0,4).Value = "Adv" 
End If  
If chkLunch = True Then  
   ActiveCell.Offset(0,5).Value = "Yes" 
Else  
   ActiveCell.Offset(0,5).Value = "No" 
End If  
If chkVegetarian = True Then  
ActiveCell.Offset(0,6).Value = "Yes" 
Else  
    If chkLunch = False Then  
    ActiveCell.Offset(0,6).Value = "" 
Else  
    ActiveCell.Offset(0,6).Value = "No" 
   End If 
End If  
Range("A1").Select 
End Sub  

 

როგორ მუშაობს CmdOK_Click() პროცედურა:  

პირველი ორი სტრიქონი უზრუნველყოფენ საჭირო ფურცლის გააქტიურებასა და მასში A1 
უჯრედის მონიშვნას:  

ActiveWorkbook.Sheets("Course Bookings").Activate 

Range("A1").Select  

კოდის Do ... Loop Until ნაწილი უზრუნველყოფს A სვეტში ცარიელი უჯრედის მოძებნას:  



Do  
  If IsEmpty(ActiveCell) = False Then  
    ActiveCell.Offset(1, 0).Select 
  End If  
Loop Until IsEmpty(ActiveCell) = True  
 
შემდეგი კოდის ოთხი სტრიქონი უზრუნველყოფს ფორმის მონაცემების ჩაწერას Excel–ის 
დოკუმენტის ფურცელში:  

ActiveCell.Value = txtName.Value  

ActiveCell.Offset(0, 1) = txtPhone.Value 

ActiveCell.Offset(0, 2) = cboDepartment.Value 

ActiveCell.Offset(0, 3) = cboCourse.Value  

 

შემდეგ ხდება ოფციური ღილაკების დამუშავება:  

      If optIntroduction = True Then  
        ActiveCell.Offset(0, 4).Value = "Intro" 
      ElseIf optIntermediate = True Then  
        ActiveCell.Offset(0, 4).Value = "Intermed" 

  Else  

    ActiveCell.Offset(0, 4).Value = "Adv"  

  End If  

 

შემდეგ უნდა მოხდეს თოლიების (check box) დამუშავება:  

   If chkLunch = True Then  
     ActiveCell.Offset(0, 5).Value = "Yes" 
   Else  
     ActiveCell.Offset(0, 5).Value = "No" 
   End If  
 

თოლია Vegetarian ელემენტისათვის უნდა დამუშავდეს შემდეგნაირად: თუ ის არჩეულია, მაშინ 
Excel–ის დოკუმენტის ფურცელში გაკეთდება ჩანაწერი "Yes";  თუ არ არის არჩეული, მაშინ:  თუ 
Lunch ელემენტი არ არის არჩეული ჩანაწერი არ გაკეთდება, თუ Lunch ელემენტი არჩეულია, მაშინ 
გაკეთდება ჩანაწერი "No":  

    If chkVegetarian = True Then 
       ActiveCell.Offset(0, 6).Value = "Yes" 
    Else  
       If chkLunch = False Then  
         ActiveCell.Offset(0, 6).Value = "" 
       Else  
         ActiveCell.Offset(0, 6).Value = "No" 



       End If  
    End If  
 

და ბოლოს მოხდება საწყის მდგომარეობაზე დაბრუნება:  

Range("A1").Select  

 

საბოლოოდ დაგვჭირდება კიდევ ერთი პროცედურა: 
 
Private Sub chkLunch_Change() 
    If chkLunch = True Then 
        chkVegetarian.Enabled = True 
    Else 
        chkVegetarian.Enabled = False 
        chkVegetarian = False 
    End If 
End Sub 
 
ფორმის გაშვება: 

ფორმის გაშვება შეიძლება მოხდეს მარტივი მაკროთი, რომელსაც შეიძლება დაუნიშნოთ ცხელი 
კლავიში ან რაიმე პიქტოგრამა სწრაფი შეღწევის პანელში, ან ღილაკი თვითონ Excel–ის დოკუმენტის 
ფურცელში. კერძოდ, Excel–ის მენიუს ზოლში, Developer ჩანართში, Controls ჯგუფში აირჩიეთ 
Insert -> Form Controls -> Button  და მოათავსეთ ღილაკი ფურცელზე. თაგვის მარჯვენა ღილაკით 
დაწკაპუნებით გამოწვეულ კონტექსტურ მენიუში აირჩიეთ Edit Text და ღილაკს მიეცით სახელი 
Start Form. შემდეგ, კვლავ ღილაკზე მარჯვენა დაწკაპუნებით აირჩიეთ Assign Macro და გამოსულ 
ფანჯარაში აირჩიეთ New, ხოლო მიღებულ პროცედურაში ჩაწერეთ:  
 
Sub Button1_Click() 
  frmCourseBooking.Show 
End Sub 
 

 

ან კიდევ, შექმენით ახალი მოდული და ჩაწერეთ პროცედურა: 

Sub OpenCourseBookingForm() 
     frmCourseBooking.Show 
End Sub  

 

 

 



 

 

 

 

 

 

 

 

 

 

 

 



TextBox ................ Name: txtFirstName 

Label..................... Caption: First Name:  

TextBox ................ Name: txtLastName  

Label..................... Caption: Last Name:  

ComboBox ............ Name: cboDepartment  

Label..................... Caption: Department:  

TextBox ................ Name: txtDate  

Label..................... Caption: Date:  

TextBox ................ Name: txtAmount  

Label..................... Caption: Amount: 

CheckBox.............. Name: chkReceipt, Caption: Receipt?  

TextBox ................ Name: txtDescription, Height: 45, Width: 132, Scrollbars: 2-fmScrollbarsVertical 

Label..................... Caption: Description:  

CommandButton ... Name: cmdOK, Caption: OK  

CommandButton ... Name: cmdClear, Caption: Clear  

CommandButton ... Name: cmdCancel, Caption: Cancel 

 

 

 

 

 

 

 

 

 



Private Sub cmdCancel_Click() 
Unload Me 
End Sub 

 

Private Sub cmdClear_Click() 
' Clear the form 
For Each ctl In Me.Controls 
If TypeName(ctl) = "TextBox" Or TypeName(ctl) = "ComboBox" Then 
ctl.Value = "" 
ElseIf TypeName(ctl) = "CheckBox" Then 
ctl.Value = False 
End If 
Next ctl 
End Sub 
 
 
 
Private Sub cmdOK_Click() 
Dim RowCount As Long 
Dim ctl As Control 
' Check user input 
If Me.txtFirstName.Value = "" Then 
MsgBox "Please enter a First Name." 
Me.txtFirstName.SetFocus 
Exit Sub 
End If 
If Me.txtLastName.Value = "" Then 
MsgBox "Please enter a Last Name." 
Me.txtFirstName.SetFocus 
Exit Sub 
End If 
If Me.cboDepartment.Value = "" Then 
MsgBox "Please choose a Department." 
Me.txtFirstName.SetFocus 
Exit Sub 
End If 
If Me.txtDate.Value = "" Then 
MsgBox "Please enter a Date." 
Me.txtFirstName.SetFocus 
Exit Sub 
End If 
If Me.txtAmount.Value = "" Then 
MsgBox "Please enter an Amount." 
Me.txtFirstName.SetFocus 
Exit Sub 
End If 
If Me.txtDescription.Value = "" Then 



MsgBox "Please enter a Description." 
Me.txtFirstName.SetFocus 
Exit Sub 
End If 
If Not IsNumeric(Me.txtAmount.Value) Then 
MsgBox "The Amount box must contain a number." 
Me.txtAmount.SetFocus 
Exit Sub 
End If 
If Not IsDate(Me.txtDate.Value) Then 
MsgBox "The Date box must contain a date." 
Me.txtDate.SetFocus 
Exit Sub 
End If 
' Write data to worksheet 
RowCount = Worksheets("Sheet1").Range("A1").CurrentRegion.Rows.Count 
With Worksheets("Sheet1").Range("A1") 
.Offset(RowCount, 0).Value = Me.txtFirstName.Value 
.Offset(RowCount, 1).Value = Me.txtLastName.Value 
.Offset(RowCount, 2).Value = Me.cboDepartment.Value 
.Offset(RowCount, 3).Value = DateValue(Me.txtDate.Value) 
.Offset(RowCount, 4).Value = Me.txtAmount.Value 
.Offset(RowCount, 5).Value = Me.txtDescription.Value 
.Offset(RowCount, 6).Value = Format(Now, "dd/mm/yyyy hh:nn:ss") 
If Me.chkReceipt.Value = True Then 
.Offset(RowCount, 7).Value = "Yes" 
Else 
.Offset(RowCount, 7).Value = "No" 
End If 
End With 
' Clear the form 
For Each ctl In Me.Controls 
If TypeName(ctl) = "TextBox" Or TypeName(ctl) = "ComboBox" Then 
ctl.Value = "" 
ElseIf TypeName(ctl) = "CheckBox" Then 
ctl.Value = False 
End If 
Next ctl 
End Sub 



ToggleButton control 

A ToggleButton control has the same appearance as a CommandButton control until you click it. When you 

click a ToggleButton control, it appears to be pressed or pushed down. The Value property of a 

ToggleButton control is True when the button is selected and False when the button is not selected. If the 

value of the TripleState property is True, a ToggleButton control can also have a value of Null. A 

ToggleButton control that has a value of Null appears to be unavailable.  

 

Note The On value, the Yes value, and the True value indicate that a ToggleButton control is selected. The 

Off value, the No value, and the False value indicate that a ToggleButton control is not selected.  

How to obtain the value of a ToggleButton control 

To obtain the value of a ToggleButton control, follow these steps:  

1. Start Excel, and then open a new blank workbook.  

2. On the Tools menu, point to Macro, and then click Visual Basic Editor.  

3. On the Insert menu, click UserForm to insert a UserForm in your workbook.  

4. Add a ToggleButton control on the UserForm.  

5. Add a Label control to the UserForm.  

6. Double-click the ToggleButton control to open the Code window for the ToggleButton control.  

7. In the Code window, type the following code for the ToggleButton1Click event:  

Private Sub ToggleButton1_Click() 
 
    If ToggleButton1.Value = True Then 
        ' Set UserForm background to Red. 
        Me.BackColor = RGB(255, 0, 0) 
    Else 
        ' Set UserForm background to Blue. 
        Me.BackColor = RGB(0, 0, 255) 
    End If 
 
End Sub 

8. On the Run menu, click Run Sub/UserForm. 

When you click the ToggleButton control, the background color of the UserForm changes.  

 

 

 

 



TabStrip control 

Use a TabStrip control to view different sets of information for a set of controls.  

How to control a TabStrip control programmatically 

To change the BackColor property of an Image control based on the tab that is selected, follow these steps:  

1. Start Excel, and then open a new blank workbook.  

2. On the Tools menu, point to Macro, and then click Visual Basic Editor.  

3. On the Insert menu, click UserForm to insert a UserForm in your workbook.  

4. Add a TabStrip control to the UserForm.  

5. Add an Image control that covers the base of the TabStrip control, but that does not cover the 

tabs.  

6. In the Properties pane for Image1, type &H000000FF& in the BackColor property.  

7. Double-click the TabStrip control to open the Code window for the TabStrip control.  

8. In the Code window, type the following code for the TabStrip1 Change event:  

Private Sub TabStrip1_Change() 
    Dim i As Integer 
    i = TabStrip1.SelectedItem.Index 
    Select Case i 
        Case 0 
            ' If Tab1 is selected, change the color of Image control to Red. 
            Image1.BackColor = RGB(255, 0, 0) 
        Case 1 
            ' If Tab2 is selected, change the color of Image control to Green. 
            Image1.BackColor = RGB(0, 255, 0) 
    End Select 
End Sub 

9. On the Run menu, click Run Sub/UserForm. 

The color of the Image control changes depending on the page in the TabStrip control that is active. 

 

 

MultiPage control 

Use a MultiPage control to work with a lot of information that can be sorted into several categories. A 

MultiPage control is made up of one or more Page objects that each contain a different set of controls. You 

can set the active page programmatically by setting the Value property of the MultiPage control.  

How to control a MultiPage control programmatically 

To add a MultiPage control and control it by using a macro, follow these steps:  



1. Start Excel, and then open a new blank workbook.  

2. On the Tools menu, point to Macro, and then click Visual Basic Editor.  

3. On the Insert menu, click UserForm to insert a UserForm in your workbook.  

4. Add a MultiPage control to the UserForm.  

5. Add a Label control to Page1 on the MultiPage control.  

6. Add a TextBox control to Page1 on the MultiPage control.  

7. On the MultiPage control, click Page2, and then repeat steps 5 and 6 to add a Label control and a 

TextBox control.  

8. Double-click the MultiPage control to open the Code window for the MultiPage control.  

9. In the Code window, type the following code for the MultiPage1 Change event:  

Private Sub MultiPage1_Change() 
 
    Select Case MultiPage1.Value 
        ' If activating Page1... 
        Case 0 
            Label1.Caption = TextBox2.Text 
            TextBox1.Text = "" 
        ' If activating Page2... 
        Case 1 
            Label2.Caption = TextBox1.Text 
            TextBox2.Text = "" 
    End Select 
 
End Sub 

10. In the Code window, type the following code for the UserForm Initialize event:  

Private Sub UserForm_Initialize() 
 
    ' Force Page1 to be active when UserForm is displayed. 
    MultiPage1.Value = 0 
    Label1.Caption = "" 
 
End Sub 

11. On the Run menu, click Run Sub/UserForm. 

 

In the TextBox control on Page1, type Test. When you click the Page2 tab, TextBox2 is cleared, 

and the caption of Label2 changes to the entry that you made in TextBox1 on Page1 ("Test"). 

 

 



How to create a wizard interface by using a MultiPage control 

When a task requires several incremental steps, a wizard interface can be very effective. You can use the 

MultiPage control to create a wizard interface instead of using multiple UserForms. This example manipulates a 

MultiPage control that has three pages. A procedure that is attached to the Initialize event of the UserForm 

disables Page2 and Page3, and forces Page1 of the MultiPage control to be active. 

 

Note When you index the pages of a MultiPage control by using the Pages collection, the first page in the 

collection is page zero. This procedure also sets the caption of the CommandButton controls and disables the 

<Back button.  

 

Note The procedure that is assigned to the Click event of CommandButton1 controls the functionality of the 

<Back button. The procedure that is assigned to the Click event of CommandButton2 controls the 

functionality of the Next> button. To create a wizard interface by using a MultiPage control, follow these 

steps:  

1. Start Excel, and then open a new blank workbook.  

2. On the Tools menu, point to Macro, and then click Visual Basic Editor.  

3. On the Insert menu, click UserForm to insert a UserForm in your workbook.  

4. Add a MultiPage control to the UserForm.  

5. Right-click the Page1 tab, and then click New Page to add Page3 to the MultiPage control.  

6. Add a CommandButton control on the UserForm that is not on the MultiPage control.  

 

Repeat this step to add a second CommandButton control on the UserForm.  

7. Double-click the UserForm to open the Code window for the UserForm.  

8. In the Code window, type the following code for the UserForm Initialize event:  

Private Sub UserForm_Initialize() 
 
    With MultiPage1 
        ' The next 2 lines disable Page2 & Page3. 
        .Pages(1).Enabled = False 
        .Pages(2).Enabled = False 
        ' Make Page1 the active page. 
        .Value = 0 
    End With 
     
    ' Set the caption on the CommandButtons. 
    CommandButton1.Caption = "<Back" 
    CommandButton1.Enabled = False 
    CommandButton2.Caption = "Next>" 
 
End Sub 



 
' Procedure for the "<Back" button 
Private Sub CommandButton1_Click() 
    Select Case MultiPage1.Value 
        Case 1                                 ' If Page2 is active... 
            With MultiPage1 
                .Pages(0).Enabled = True       ' Enable Page1. 
                .Value = MultiPage1.Value ‐ 1  ' Move back 1 page. 
                .Pages(1).Enabled = False      ' Disable Page2. 
            End With 
            CommandButton1.Enabled = False     ' Disable Back button. 
             
        Case 2                                 ' If Page3 is active... 
            With MultiPage1 
                .Pages(1).Enabled = True       ' Enable Page2. 
                .Value = MultiPage1.Value ‐ 1  ' Move back 1 page. 
                .Pages(2).Enabled = False      ' Disable Page3. 
   CommandButton2.Caption = "Next>" 
            End With 
    End Select 
 
End Sub 
 
' Procedure for the "Next>" button 
Private Sub CommandButton2_Click() 
 
    Select Case MultiPage1.Value 
        Case 0                                  ' If Page1 is active... 
            With MultiPage1 
                .Value = MultiPage1.Value + 1   ' Move forward 1 page. 
                .Pages(1).Enabled = True        ' Enable Page2. 
                .Pages(0).Enabled = False       ' Disable Page1. 
            End With 
            CommandButton1.Enabled = True       ' Enable Back button. 
             
        Case 1                                  ' If Page2 is active... 
            With MultiPage1 
                .Value = MultiPage1.Value + 1   ' Move forward 1 page. 
                .Pages(2).Enabled = True        ' Enable Page3. 
                .Pages(1).Enabled = False       ' Disable Page2. 
            End With 
            CommandButton2.Caption = "Finish"   ' Change Next button to Finish. 
             
        Case 2                                  ' If Page3 is active... 
            MsgBox "Finished!"                  ' User is Finished. 
            Unload Me                           ' Unload the UserForm. 
    End Select 
 
End Sub 



9. On the Run menu, click Run Sub/UserForm. 

When you click Next>, Page2 is activated and the <Back button becomes available. When you click Next> a 

second time, Page3 is activated and the caption for CommandButton2 changes to "Finish".  

 

 

ScrollBar control 

You can use a ScrollBar control when you want to change the value that is displayed by another control, such 

as a Label control.  

How to change a Label control that is based on the value of a ScrollBar control 

To change the Caption property of a Label control to the current setting of the Value property of a ScrollBar 

control, follow these steps:  

1. Start Excel, and then open a new blank workbook.  

2. On the Tools menu, point to Macro, and then click Visual Basic Editor.  

3. On the Insert menu, click UserForm to insert a UserForm in your workbook.  

4. Add a ScrollBar control to the UserForm.  

5. Add a Label control to the UserForm.  

6. Double-click the ScrollBar control to open the Code window for the ScrollBar control.  

7. In the Code window, type the following code for the ScrollBar1 Change event:  

Private Sub ScrollBar1_Change() 
 
    Label1.Caption = ScrollBar1.Value 
 
End Sub 

8. On the Run menu, click Run Sub/UserForm. 

When you scroll by using the ScrollBar control, Label1 is updated with the current value of the ScrollBar 

control.  

 

 

 

 

 



SpinButton control 

A SpinButton control, like a ScrollBar control, is frequently used to increment or to decrement the value of 

another control, such as a Label control. The SmallChange property determines how much the value of a 

SpinButton control changes when it is clicked.  

How to add a SpinButton control that increments or decrements a date that is stored in a 

TextBox control 

To add a SpinButton control that increments or decrements a date that is stored in a TextBox control, follow 

these steps:  

1. Start Excel, and then open a new blank workbook.  

2. On the Tools menu, point to Macro, and then click Visual Basic Editor.  

3. On the Insert menu, click UserForm to insert a UserForm in your workbook.  

4. Add a SpinButton control to the UserForm.  

5. Add a TextBox control to the UserForm.  

6. Double-click the SpinButton control to open the Code window for the SpinButton control.  

7. In the Code window, type the following code for the SpinButton1 SpinUp event:  

Private Sub SpinButton1_SpinUp() 
    TextBox1.Text = DateValue(TextBox1.Text) + 1 
End Sub 

8. In the Code window, type the following code for the SpinButton1 SpinDown event:  

Private Sub SpinButton1_SpinDown() 
    TextBox1.Text = DateValue(TextBox1.Text) ‐ 1 
End Sub 

9. In the Code window, type the following code for the UserForm Initialize event:  

Private Sub UserForm_Initialize() 
    TextBox1.Text = Date 
End Sub 

10. On the Run menu, click Run Sub/UserForm. 

When the UserForm appears, the current date appears in TextBox1. When you click the SpinButton control, 

the date is incremented or decremented by one day. 

 

In this example, if you change the SmallChange property of SpinButton1, you do not affect the number of 

days the entry in TextBox1 is changed by when you click SpinButton1. The number of days is determined 



only by the procedure that you attached to the SpinUp event and the SpinDown event of SpinButton1. 

 

RefEdit control 

The RefEdit control imitates the behavior of the reference boxes that are built into Excel. You can use the 

Value property to obtain the current cell address that are stored in a RefEdit control.  

How to populate a range of cells based on the range that you select by using the RefEdit 

control 

To use the RefEdit control to populate cells, follow these steps:  

1. Start Excel, and then open a new blank workbook.  

2. On the Tools menu, point to Macro, and then click Visual Basic Editor.  

3. On the Insert menu, click UserForm to insert a UserForm in your workbook.  

4. Add a RefEdit control to the UserForm.  

5. Add a CommandButton control to the UserForm.  

6. Double-click the CommandButton control to open the Code window for the CommandButton 

control.  

7. In the Code window, type the following code for the CommandButton1 Click event:  

Private Sub CommandButton1_Click() 
 
    Dim MyRange As String 
    MyRange = RefEdit1.Value 
    Range(MyRange).Value = "test" 
    Unload Me 
 
End Sub 

8. On the Run menu, click Run Sub/UserForm. 

The UserForm appears.  

9. Click the button in the RefEdit control.  

Notice that the UserForm collapses.  

10. Select a range of cells such as A1:A5, and then click the button in the RefEdit control to expand the 

UserForm.  

11. Click CommandButton1. 

The UserForm closes and the cells that you selected now contain the word "test". 

 

 

 



Image control 

The purpose of the Image control is to display a picture on a UserForm. To assign a picture to an Image 

control at run time, use the LoadPicture function.  

How to load a picture into an Image control 

To insert an Image control that prompts you to select a picture to load when you click the Image control, 

follow these steps:  

1. Start Excel, and then open a new blank workbook.  

2. On the Tools menu, point to Macro, and then click Visual Basic Editor.  

3. On the Insert menu, click UserForm to insert a UserForm in your workbook.  

4. Add an Image control on the UserForm.  

5. Double-click the Image control to open the Code window for the Image control.  

6. In the Code window, type the following code for the Image1 Click event:  

Private Sub Image1_Click() 
 
    Dim fname As String 
     
    ' Display the Open dialog box. 
    fname = Application.GetOpenFilename(filefilter:= _ 
            "Bitmap Files(*.bmp),*.bmp", Title:="Select Image To Open") 
     
    ' If you did not click Cancel... 
    If fname <> "False" Then 
 
        ' Load the bitmap into the Image control. 
        Image1.Picture = LoadPicture(fname) 
 
        ' Refresh the UserForm. 
        Me.Repaint 
    End If 
 
End Sub 

7. On the Run menu, click Run Sub/UserForm.  

 

The UserForm appears.  

8. Click the Image control.  

 

When you click the Image control, the Select Image To Open dialog box appears, and then you 

can select a bitmap file to insert into the control. 

